



**HAL**  
open science

## A locally nameless solution to the POPLmark challenge

Xavier Leroy

► **To cite this version:**

Xavier Leroy. A locally nameless solution to the POPLmark challenge. [Research Report] 2007, pp.54.  
inria-00123945v1

**HAL Id: inria-00123945**

**<https://inria.hal.science/inria-00123945v1>**

Submitted on 11 Jan 2007 (v1), last revised 11 Jan 2007 (v2)

**HAL** is a multi-disciplinary open access archive for the deposit and dissemination of scientific research documents, whether they are published or not. The documents may come from teaching and research institutions in France or abroad, or from public or private research centers.

L'archive ouverte pluridisciplinaire **HAL**, est destinée au dépôt et à la diffusion de documents scientifiques de niveau recherche, publiés ou non, émanant des établissements d'enseignement et de recherche français ou étrangers, des laboratoires publics ou privés.



INSTITUT NATIONAL DE RECHERCHE EN INFORMATIQUE ET EN AUTOMATIQUE

*A locally nameless solution  
to the POPLmark challenge*

Xavier Leroy

N° ????

Janvier 2007

Thème SYM

*R*apport  
de recherche





## A locally nameless solution to the POPLmark challenge

Xavier Leroy

Thème SYM — Systèmes symboliques  
Projet Gallium

Rapport de recherche n° ??? — Janvier 2007 — 54 pages

**Abstract:** The POPLmark challenge is a collective experiment intended to assess the usability of theorem provers and proof assistants in the context of fundamental research on programming languages. In this report, we present a solution to the challenge, developed with the Coq proof assistant, and using the “locally nameless” presentation of terms with binders introduced by McKinna, Pollack, Gordon, and McBride.

**Key-words:** POPLmark, Coq, locally nameless, alpha-conversion, binders, type systems, metatheory, system F-sub

## Une solution sans noms locaux à l'expérience POPLmark

**Résumé :** L'expérience collective POPLmark vise à évaluer l'utilisabilité des démonstrateurs automatiques et des assistants de preuves dans le contexte de la recherche fondamentale sur les langages de programmation. Dans ce rapport, nous présentons une solution à cette expérience, développée à l'aide de l'assistant de preuves Coq, et reposant sur la présentation "sans noms locaux" des termes avec lieurs introduite par McKinna, Pollack, Gordon, et McBride.

**Mots-clés :** POPLmark, Coq, sans noms locaux, alpha-conversion, lieurs, systèmes de types, métathéorie, système F-sub

# Chapter 1

## Introduction

### 1.1 The POPLmark challenge

The POPLmark challenge [ABF<sup>+</sup>05] is a collective experiment intended to assess the usability of theorem provers and proof assistants in the context of fundamental research on programming languages. The need for computer assistance when formalizing and proving properties of programming languages (formal semantics and type systems) is well expressed in the statement of the challenge:

Many proofs about programming languages are long, straightforward, and tedious, with just a few interesting cases. Their complexity arises from the management of many details rather than from deep conceptual difficulties; yet small mistakes or overlooked cases can invalidate large amounts of work. These effects are amplified as languages scale: it becomes hard to keep definitions and proofs consistent, to reuse work, and to ensure tight relationships between theory and implementations. Automated proof assistants offer the hope of significantly easing these problems. However, despite much encouraging progress in recent years and the availability of several mature tools, their use is still not commonplace.

The challenge itself consists of formalizing the operational semantics and the type system of  $F_{<}$ , a typed functional language featuring polymorphism and subtyping, and proving the soundness of the type system with respect to this semantics. As a guidance, the statement of the challenge provides a detailed on-paper formalization of  $F_{<}$  and a proof of type soundness written in ordinary mathematics. The challenge itself is to express these formalization and proofs in a theorem prover in such a way that the proofs can be mechanically checked.

A dozen complete or partial solutions to the POPLmark challenge were developed, using a wide variety of proof assistants (Coq, Isabelle/HOL, Twelf, ...) and of representation techniques for the terms, types and rules of the  $F_{<}$  language. These solutions can be found

on the POPLmark Web site<sup>1</sup>. This report presents the solution that we developed. It uses the Coq proof assistant [Coq07, BC04], as well as the so-called “locally nameless” representation introduced by McKinna, Pollack, Gordon, and McBride, [MP99, Gor94, MM04]. The complete Coq development is available online [Ler07].

## 1.2 Locally nameless representations

Perhaps surprisingly, the main difficulty in the POPLmark challenge is not to translate the high-level, on-paper proofs into equivalent mechanized proofs. These on-paper proofs are mostly syntactic in nature and do not involve higher mathematics. Such proofs are handled well by many existing proof assistants.

What makes POPLmark difficult is the need to correctly handle *binders*, *bound variables* and *alpha-conversion*. The terms and types of  $F_{<}$  contain variables such as function parameters and type variables, as well as constructs that bind these variables, such as function abstractions  $\lambda x. x$  in terms and universal quantification  $\forall X. X \rightarrow X$  in types. To obtain a sound theory of the language, it is necessary to treat terms and types as equal up to alpha-conversion, that is, renamings of bound variables. For instance, the two types  $\forall X. X \rightarrow X$  and  $\forall Y. Y \rightarrow Y$  must be treated as equal. This is difficult to achieve with today’s proof assistants, because these offer poor support for quotient sets. Several representations of binders have been investigated over the last 40 years to overcome this difficulty.

**Nominal representations** follow the usual mathematical practice of identifying bound variables by names and quotienting terms up to alpha-conversion of bound names. This can be internalized within the logic itself, leading to nominal logics [Pit03]. Support for a nominal logic within a proof assistant is still in its infancy, but is making significant progress [UT05]. Despite these difficulties, this approach is attractive because it is the closest to usual mathematical practice, leading to statements and proofs that are very close to what we are accustomed to see in textbooks and research papers.

**de Bruijn indices** avoid the issue with alpha-conversion by representing variables not by names, but by position (indices) relative to the enclosing binders:  $v_1$  is the variable bound by the first enclosing binder,  $v_2$  by the second, etc. For instance, the term  $\lambda x. x (\lambda y. x y)$  in nominal notation is represented as  $\lambda. v_1 (\lambda. v_2 v_1)$  in de Bruijn notation. The strength of this approach is that terms have unique representations: two terms are alpha-equivalent in nominal notation if and only if they are identical in de Bruijn notation. It is therefore easy to represent and work with in a proof assistant [Hue94]. The downside of this approach is that the meaning of de Bruijn indices is very dependent on the context. Many lifting and relocation operations over indices must be included in the statements of theorems, making them unnatural and hard to read.

<sup>1</sup><http://fling-1.seas.upenn.edu/~plclub/cgi-bin/poplmark/>

**Higher-order abstract syntax (HOAS)** uses the functions provided by the logic to represent binders. For example, we have the following representations for the term  $\lambda x.x$ :

Style	Representation of $\lambda x.x$
nominal	<code>Lambda("x", Var "x")</code>
de Bruijn	<code>Lambda(Var 1)</code>
HOAS	<code>Lambda(fun x -&gt; x)</code>

(The `fun` construct denotes a function of the logic, binding a logical variable `x` that ranges over all terms.) The beauty of this approach is that alpha-conversion and substitution of bound names are handled automatically by the logic and need not be managed explicitly. The downside of HOAS is that it is not compatible with the rich logics of general-purpose proof assistants such as Coq and Isabelle/HOL and therefore can only be supported by systems such as Twelf’s metatheory, which are less expressive. Also, it leads to statements of theorems that are quite different from what we write in ordinary mathematics.

In our solution to the POPLmark challenge, we use a representation for binders that is a combination of the nominal approach with de Bruijn indices. This representation is known as the “locally nameless” approach [MP99, Gor94, MM04]. It uses de Bruijn indices to identify bound variables, and names to identify free variables. For example, we have the following representations:

Style	Representation of $\lambda x. y x$
nominal	<code>Lambda("x", App(Var "y", Var "x"))</code>
de Bruijn	<code>Lambda(App(Var 2, Var 1))</code>
locally nameless	<code>Lambda(App(Freevar "y", Boundvar 1))</code>

Like de Bruijn indices, the locally nameless representation has the very nice property that alpha-equivalent terms have syntactically equal representations. Like nominal approaches, the locally nameless representation enables us to use familiar names to refer to free variables, leading to statements of theorems that are close to ordinary mathematics.

A crucial invariant in the locally nameless approach is that representations of terms never contain free de Bruijn indices. Consequently, when recursing over a de Bruijn-closed term and encountering a binder such as `Lambda(t)`, it is incorrect to recurse over `t`, since this term can contain `Boundvar 1` as a free de Bruijn index. The correct approach is to invent a fresh name `x` to stand for the formal parameter of the lambda-abstraction and recurse over `t[0 ← x]`, that is, the term obtained by substituting the free variable `Freevar "x"` for all occurrences of the bound variable `Boundvar 1` in `t`. (See section 2.2.3 for detailed explanations.) This style of definition is slightly unnatural at first, but is reasonably easy to get used to. In particular, this substitution by a fresh name materializes the so-called “Barendregt convention” for nominal terms.



### 1.3 Outline

The remainder of this report is a complete step-by-step presentation of our Coq development in literate programming style. All definitions, theorems and intermediate lemmas are shown in Coq syntax, interspersed with explanations. We omit the proof scripts for all lemmas but show them for the main theorems, so that the reader can get a feeling of the size and complexity of the proof scripts. The full Coq development, including scripts, is available online [Ler07].

- Chapter 2 defines the syntax of  $F_{<}$ : type expressions and the subtyping relation between types, and proves three key properties of this relation: reflexivity, transitivity, and stability by substitution. It corresponds to part 1A and a fragment of part 2A of the POPLmark challenge.
- Chapter 3 defines the syntax of terms, the type system, and the dynamic semantics of  $F_{<}$ . It proves the soundness of the type system with respect to the semantics. This corresponds to the remainder of part 2A of the challenge.
- Chapter 4 shows how executions of  $F_{<}$ : programs can be performed (for testing purposes), both within Coq and through “extraction” (automatic generation) of Caml code for a reference  $F_{<}$ : interpreter. This is part 3 of the challenge.
- Chapter 5 concludes this report by an informal assessment of the quality of our solution, and more generally of the usability of the “locally nameless” approach and of the Coq proof assistant for POPLmark-style problems.

## Chapter 2

# Algorithmic subtyping

This chapter corresponds to part 1A of the POPLmark challenge, namely the formal definition of a subtyping relation between types of  $F_{<}$  and the proof of basic type-theoretic properties of this relation.

We start by “importing” three modules from the Coq library of standard definitions and theorems: *Arith* (arithmetic over natural numbers), *ZArith* (arithmetic over integers) and *List* (operations over finite lists). We also import a module *extralibrary* that we developed specially, which provides additional lemmas about list membership. This module is omitted in this report, but available as part of the Web version of this development [Ler07].

```
Require Import Arith.  
Require Import ZArith.  
Require Import List.  
Require Import extralibrary.
```

### 2.1 Names and swaps of names

We use names (also called atoms) to represent free variables in terms. Any infinite type with decidable equality will do. In preparation for the second part of the challenge, we attach a kind to every name: either “type” or “term”, and ensure that there are infinitely many names of each kind. Concretely, we represent names by pairs of a kind and an integer (type  $\mathbb{Z}$ ).

```
Inductive name_kind : Set :=  
  | TYPE: name_kind  
  | TERM: name_kind.
```

Definition *name* : Set := (name\_kind  $\times$  Z)%type.

Definition *kind* (n: name) : name\_kind := fst n.

Equality between names is decidable.

Lemma *eq\_name*:  $\forall (n1\ n2: \text{name}), \{n1 = n2\} + \{n1 \neq n2\}$ .

Moreover, we have the following obvious simplification rules on tests over name equality.

Lemma *eq\_name\_true*:

$\forall (A: \text{Set}) (n: \text{name}) (a\ b: A), (\text{if } \text{eq\_name } n\ n \text{ then } a \text{ else } b) = a$ .

Lemma *eq\_name\_false*:

$\forall (A: \text{Set}) (n\ m: \text{name}) (a\ b: A), n \neq m \rightarrow (\text{if } \text{eq\_name } n\ m \text{ then } a \text{ else } b) = b$ .

The following lemma shows that there always exists a name of the given kind that is fresh w.r.t. the given list of names, that is, distinct from all the names in this list.

Definition *find\_fresh\_name* ( $k: \text{name\_kind}$ ) ( $l: \text{list name}$ ):  $\text{name} :=$   
 $(k, 1 + \text{fold\_right } (\text{fun } (n: \text{name})\ x \Rightarrow \text{Zmax } (\text{snd } n)\ x)\ 0\ l)\%Z$ .

Lemma *find\_fresh\_name\_is\_fresh*:

$\forall k\ l$ , let  $n := \text{find\_fresh\_name } k\ l$  in  $\neg \text{In } n\ l \wedge \text{kind } n = k$ .

Lemma *fresh\_name*:

$\forall (k: \text{name\_kind}) (l: \text{list name}), \exists n, \neg \text{In } n\ l \wedge \text{kind } n = k$ .

As argued by Pitts and others, swaps (permutations of two names) are an interesting special case of renamings. We will use swaps later to prove that our definitions are equivariant, that is, insensitive to the choices of fresh identifiers.

Definition *swap* ( $u\ v\ x: \text{name}$ ):  $\text{name} :=$

$\text{if } \text{eq\_name } x\ u \text{ then } v \text{ else if } \text{eq\_name } x\ v \text{ then } u \text{ else } x$ .

The following lemmas are standard properties of swaps: self-inverse, injective, kind-preserving.

Lemma *swap\_left*:  $\forall x\ y, \text{swap } x\ y\ x = y$ .

Lemma *swap\_other*:  $\forall x\ y\ z, z \neq x \rightarrow z \neq y \rightarrow \text{swap } x\ y\ z = z$ .

Lemma *swap\_inj*:  $\forall u\ v\ x\ y, \text{swap } u\ v\ x = \text{swap } u\ v\ y \rightarrow x = y$ .

Lemma *swap\_kind*:  $\forall u\ v\ x, \text{kind } u = \text{kind } v \rightarrow \text{kind } (\text{swap } u\ v\ x) = \text{kind } x$ .

## 2.2 Types and typing environments

### 2.2.1 Type expressions

The syntax of type expressions is standard, except that we have two representations for variables: *Tparam* represents free type variables, identified by a name, while *Tvar* represents

bound type variables, identified by their de Bruijn indices. Our de Bruijn indices start at 0. In a *Forall*  $t1\ t2$  type, the variable *Tvar* 0 is bound by the *Forall* in type  $t2$ .

```
Inductive type: Set :=
| Tparam: name → type
| Tvar: nat → type
| Top: type
| Arrow: type → type → type
| Forall: type → type → type.
```

The free names of a type are defined as follow. Notice the *Forall* case: *Forall* does not bind any name.

```
Fixpoint fv_type (t: type) : list name :=
  match t with
  | Tparam x ⇒ x :: nil
  | Tvar n ⇒ nil
  | Top ⇒ nil
  | Arrow t1 t2 ⇒ fv_type t1 ++ fv_type t2
  | Forall t1 t2 ⇒ fv_type t1 ++ fv_type t2
  end.
```

There are two substitution operations over types, written *vsubst* and *psubst* in Pollack's talk. *vsubst* substitutes a type for a bound variable (a de Bruijn index). *psubst* substitutes a type for a free variable (a name).

The crucial observation is that variable capture cannot occur during either substitution:

- Types never contain free de Bruijn indices, since these indices are used only for representing bound variables. Therefore, *vsubst* does not need to perform lifting of de Bruijn indices in the substituted type.
- Types never bind names, only de Bruijn indices. Therefore, *psubst* never needs to perform renaming of names in the substituted term when descending below a binder.

```
Fixpoint vsubst_type (a: type) (x: nat) (b: type) {struct a} : type :=
  match a with
  | Tparam n ⇒ Tparam n
  | Tvar n ⇒
    match compare_nat n x with
    | Nat_less _ ⇒ Tvar n
    | Nat_equal _ ⇒ b
    | Nat_greater _ ⇒ Tvar (pred n)
    end
  | Top ⇒ Top
  | Arrow a1 a2 ⇒ Arrow (vsubst_type a1 x b) (vsubst_type a2 x b)
```

| *Forall* *a1 a2*  $\Rightarrow$  *Forall* (*vsubst\_type* *a1 x b*) (*vsubst\_type* *a2 (S x) b*)  
*end.*

Fixpoint *psubst\_type* (*a: type*) (*x: name*) (*b: type*) {*struct a*} : *type* :=  
*match a with*  
| *Tparam n*  $\Rightarrow$  *if eq\_name n x then b else Tparam n*  
| *Tvar n*  $\Rightarrow$  *Tvar n*  
| *Top*  $\Rightarrow$  *Top*  
| *Arrow a1 a2*  $\Rightarrow$  *Arrow* (*psubst\_type a1 x b*) (*psubst\_type a2 x b*)  
| *Forall a1 a2*  $\Rightarrow$  *Forall* (*psubst\_type a1 x b*) (*psubst\_type a2 x b*)  
*end.*

In the remainder of the development, *vsubst* is only used to replace bound variable 0 by a fresh, free variable (a name) when taking apart a *Forall* type. This operation is similar to the “freshening” operation used in Fresh ML and related systems. We call it *freshen\_type* for clarity.

Definition *freshen\_type* (*a: type*) (*x: name*) : *type* := *vsubst\_type a 0 (Tparam x)*.

Free variables and freshening play well together.

Lemma *fv\_type\_vsubst\_type*:

$\forall x a n b, \text{In } x (\text{fv\_type } a) \rightarrow \text{In } x (\text{fv\_type } (\text{vsubst\_type } a n b)).$

Lemma *fv\_type\_freshen\_type*:

$\forall x a y, \text{In } x (\text{fv\_type } a) \rightarrow \text{In } x (\text{fv\_type } (\text{freshen\_type } a y)).$

We now define swaps (permutation of names) over types and show basic properties of swaps that will be useful later.

Fixpoint *swap\_type* (*u v: name*) (*t: type*) {*struct t*} : *type* :=  
*match t with*  
| *Tparam x*  $\Rightarrow$  *Tparam (swap u v x)*  
| *Tvar n*  $\Rightarrow$  *Tvar n*  
| *Top*  $\Rightarrow$  *Top*  
| *Arrow t1 t2*  $\Rightarrow$  *Arrow* (*swap\_type u v t1*) (*swap\_type u v t2*)  
| *Forall t1 t2*  $\Rightarrow$  *Forall* (*swap\_type u v t1*) (*swap\_type u v t2*)  
*end.*

Swaps are involutions (self-inverse).

Lemma *swap\_type\_inv*:  $\forall u v t, \text{swap\_type } u v (\text{swap\_type } u v t) = t.$

Swaps of variables that do not occur free in a type leave the type unchanged.

Lemma *swap\_type\_not\_free*:

$\forall u v t, \neg \text{In } u (\text{fv\_type } t) \rightarrow \neg \text{In } v (\text{fv\_type } t) \rightarrow \text{swap\_type } u v t = t.$

Swaps commute with *vsubst* substitution and freshening.

Lemma *vsubst\_type\_swap*:

$$\forall u v a n b, \\ \text{swap\_type } u v (\text{vsubst\_type } a n b) = \text{vsubst\_type } (\text{swap\_type } u v a) n (\text{swap\_type } u v b).$$

Lemma *freshen\_type\_swap*:

$$\forall u v a x, \\ \text{swap\_type } u v (\text{freshen\_type } a x) = \text{freshen\_type } (\text{swap\_type } u v a) (\text{swap } u v x).$$

Swaps commute with the computation of free variables.

Lemma *in\_fv\_type\_swap*:

$$\forall u v x t, \text{In } x (\text{fv\_type } t) \leftrightarrow \text{In } (\text{swap } u v x) (\text{fv\_type } (\text{swap\_type } u v t)).$$

## 2.2.2 Typing environments

Typing environments are standard: lists of (name, type) pairs. Bindings are added to the left of the environment using the *cons* list operation. Thus, later bindings come first.

Definition *typenv* := list (name × type).

Definition *dom* (e: typenv) := map (@fst name type) e.

Lemma *dom\_append*:  $\forall e1 e2, \text{dom } (e1 ++ e2) = \text{dom } e1 ++ \text{dom } e2$ .

The *lookup* function returns the type associated with a name in a typing environment.

Fixpoint *lookup* (x: name) (e: typenv) {struct e} : option type :=  
 match e with  
 | nil ⇒ None  
 | (y, t) :: e' ⇒ if eq\_name x y then Some t else lookup x e'  
 end.

Lemma *lookup\_inv*:  $\forall x t e, \text{lookup } x e = \text{Some } t \rightarrow \text{In } x (\text{dom } e)$ .

Lemma *lookup\_exists*:  $\forall x e, \text{In } x (\text{dom } e) \rightarrow \exists t, \text{lookup } x e = \text{Some } t$ .

We extend swaps to typing environments, pointwise.

Fixpoint *swap\_env* (u v: name) (e: typenv) {struct e} : typenv :=  
 match e with  
 | nil ⇒ nil  
 | (x, t) :: e' ⇒ (swap u v x, swap\_type u v t) :: swap\_env u v e'  
 end.

Environment lookup commutes with swaps.

Lemma *lookup\_swap*:

$$\forall u v x e t, \text{lookup } x e = \text{Some } t \rightarrow \\ \text{lookup } (\text{swap } u v x) (\text{swap\_env } u v e) = \text{Some } (\text{swap\_type } u v t).$$

The *dom* operation commutes with swaps.

Lemma *in\_dom\_swap*:

$$\forall u v x e, \text{In } x (\text{dom } e) \leftrightarrow \text{In } (\text{swap } u v x) (\text{dom } (\text{swap\_env } u v e)).$$

### 2.2.3 Well-formedness of types and environments

A type is well-formed in a typing environment if:

- all names free in the type are of kind *TYPE*;
- all names free in the type are bound in the environment;
- it does not contain free de Bruijn variables.

We capture these conditions by the following inference rules.

Inductive *wf\_type*: *typenv*  $\rightarrow$  *type*  $\rightarrow$  *Prop* :=

- | *wf\_type\_param*:  $\forall x e,$   
 $\text{kind } x = \text{TYPE} \rightarrow \text{In } x (\text{dom } e) \rightarrow$   
 $\text{wf\_type } e (\text{Tparam } x)$
- | *wf\_type\_top*:  $\forall e,$   
 $\text{wf\_type } e \text{Top}$
- | *wf\_type\_arrow*:  $\forall e t1 t2,$   
 $\text{wf\_type } e t1 \rightarrow \text{wf\_type } e t2 \rightarrow \text{wf\_type } e (\text{Arrow } t1 t2)$
- | *wf\_type\_forall*:  $\forall e t1 t2,$   
 $\text{wf\_type } e t1 \rightarrow$   
 $(\forall x,$   
 $\text{kind } x = \text{TYPE} \rightarrow \neg \text{In } x (\text{fv\_type } t2) \rightarrow \neg \text{In } x (\text{dom } e) \rightarrow$   
 $\text{wf\_type } ((x, t1) :: e) (\text{freshen\_type } t2 x)) \rightarrow$   
 $\text{wf\_type } e (\text{Forall } t1 t2).$

The rules are straightforward, except perhaps the *wf\_type\_forall* rule. It follows a general pattern for operating over sub-terms of a binder, such as *t2* in *Forall t1 t2*. The de Bruijn variable *Tvar 0* is potentially free in *t2*. To recover a well-formed term, without free de Bruijn variables, we substitute *Tvar 0* with a fresh name *x*. Therefore, the premise for *t2* applies to *freshen\_type t2 x*.

How should *x* be chosen? As in the name-based specification, *x* must not be in the domain of *e*, otherwise the extended environment  $(x, t1) :: e$  would be ill-formed. In addition, the name *x* must not be free in *t2*, otherwise the freshening *freshen\_type t2 x* would incorrectly identify the bound, universally-quantified variable of the *Forall* types with an existing, free type variable.

How should *x* be quantified? That is, should the second premise  $\text{wf\_type } ((x, t1) :: e) (\text{freshen\_type } t2 x)$  hold for one particular name *x* not in *dom(e)*, or for all names *x* not in

$dom(e)$ ? The “for all” alternative obviously leads to a stronger induction principle: proofs that proceed by inversion or induction over an hypothesis  $wf\_type\ e$  (*Forall*  $t1\ t2$ ) can then choose any convenient  $x$  fresh for  $e$  to exploit the second premise, rather than having to cope with a fixed, earlier choice of  $x$ . Symmetrically, the “for one” alternative is more convenient for proofs that must conclude  $wf\_type\ e$  (*Forall*  $t1\ t2$ ): it suffices to exhibit one suitable  $x$  fresh in  $e$  that satisfies the second premise, rather than having to establish the second premise for all such  $x$ .

The crucial observation is that those two alternative are equivalent: the same subtyping judgements can be derived with the “for all” rule and the “for one” rule. (See Pollack’s talk for more explanations.) Therefore, in the definition of the  $wf\_type$  predicate above, we chose the “for all” rule, so as to get the strongest induction principle. And we will show shortly that the “for one” rule is admissible and can be used in proofs that conclude  $wf\_type\ e$  (*Forall*  $t1\ t2$ ).

An environment is well-formed if every type it contains is well-formed in the part of the environment that occurs to its right, i.e. the environment at the time this type was introduced. This ensures in particular that all the variables in this type are bound earlier (i.e. to the right) in the environment. Moreover, we impose that no name is bound twice in an environment.

Inductive  $wf\_env: typenv \rightarrow Prop :=$   
 |  $wf\_env\_nil:$   
    $wf\_env\ nil$   
 |  $wf\_env\_cons: \forall x\ t\ e,$   
    $wf\_env\ e \rightarrow \neg In\ x\ (dom\ e) \rightarrow wf\_type\ e\ t \rightarrow$   
    $wf\_env\ ((x, t) :: e).$

Lemma  $wf\_type\_env\_incr:$   
 $\forall e\ t, wf\_type\ e\ t \rightarrow \forall e', incl\ (dom\ e)\ (dom\ e') \rightarrow wf\_type\ e'\ t.$

A type well formed in  $e$  has all its free names in the domain of  $e$ .

Lemma  $fv\_wf\_type: \forall x\ e\ t, wf\_type\ e\ t \rightarrow In\ x\ (fv\_type\ t) \rightarrow In\ x\ (dom\ e).$

Looking up the type of a name in a well-formed environment returns a well-formed type.

Lemma  $wf\_type\_lookup: \forall e, wf\_env\ e \rightarrow \forall x\ t, lookup\ x\ e = Some\ t \rightarrow wf\_type\ e\ t.$

Type well-formedness is stable by swapping.

Lemma  $wf\_type\_swap:$   
 $\forall u\ v\ e\ t,$   
 $kind\ u = kind\ v \rightarrow wf\_type\ e\ t \rightarrow wf\_type\ (swap\_env\ u\ v\ e)\ (swap\_type\ u\ v\ t).$

Environment well-formedness is stable by swapping.

Lemma  $wf\_env\_swap:$   
 $\forall u\ v\ e, kind\ u = kind\ v \rightarrow wf\_env\ e \rightarrow wf\_env\ (swap\_env\ u\ v\ e).$



The domain of an environment is invariant by swaps of names that are not in this domain.

Lemma *swap\_env\_dom*:

$$\forall u v e, \neg \text{In } u \text{ (dom } e) \rightarrow \neg \text{In } v \text{ (dom } e) \rightarrow \text{dom (swap\_env } u v e) = \text{dom } e.$$

A well-formed environment is invariant by swaps of names that are not in the domains of this environment.

Lemma *swap\_env\_not\_free*:

$$\forall u v e, \text{wf\_env } e \rightarrow \neg \text{In } u \text{ (dom } e) \rightarrow \neg \text{In } v \text{ (dom } e) \rightarrow \text{swap\_env } u v e = e.$$

We now show that the alternate formulation of rule *wf\_type\_forall* (the one with “for one fresh name  $x$ ” instead of “for all fresh names  $x$ ” in the second premise) is admissible.

Lemma *wf\_type\_forall'*:

$$\begin{aligned} & \forall e x t t1 t2, \\ & \text{wf\_type } e t1 \rightarrow \text{kind } x = \text{TYPE} \rightarrow \neg \text{In } x \text{ (fv\_type } t2) \rightarrow \neg \text{In } x \text{ (dom } e) \rightarrow \\ & \text{wf\_type } ((x, t) :: e) \text{ (freshen\_type } t2 x) \rightarrow \\ & \text{wf\_type } e \text{ (Forall } t1 t2). \end{aligned}$$

## 2.3 Algorithmic subtyping

We now define the subtyping judgement as an inductive predicate. Each constructor of the predicate corresponds to an inference rule in the original definition of subtyping.

Inductive *is\_subtype*: *typenv*  $\rightarrow$  *type*  $\rightarrow$  *type*  $\rightarrow$  *Prop* :=

$$\begin{aligned} & | \text{sa\_top}: \forall e s, \\ & \quad \text{wf\_env } e \rightarrow \text{wf\_type } e s \rightarrow \\ & \quad \text{is\_subtype } e s \text{ Top} \\ & | \text{sa\_refl\_tvar}: \forall e x u, \\ & \quad \text{wf\_env } e \rightarrow \text{kind } x = \text{TYPE} \rightarrow \text{lookup } x e = \text{Some } u \rightarrow \\ & \quad \text{is\_subtype } e \text{ (Tparam } x) \text{ (Tparam } x) \\ & | \text{sa\_trans\_tvar}: \forall e x u t, \\ & \quad \text{kind } x = \text{TYPE} \rightarrow \text{lookup } x e = \text{Some } u \rightarrow \text{is\_subtype } e u t \rightarrow \\ & \quad \text{is\_subtype } e \text{ (Tparam } x) t \\ & | \text{sa\_arrow}: \forall e s1 s2 t1 t2, \\ & \quad \text{is\_subtype } e t1 s1 \rightarrow \text{is\_subtype } e s2 t2 \rightarrow \\ & \quad \text{is\_subtype } e \text{ (Arrow } s1 s2) \text{ (Arrow } t1 t2) \\ & | \text{sa\_all}: \forall e s1 s2 t1 t2, \\ & \quad \text{is\_subtype } e t1 s1 \rightarrow \\ & \quad (\forall x, \\ & \quad \quad \text{kind } x = \text{TYPE} \rightarrow \neg \text{In } x \text{ (dom } e) \rightarrow \\ & \quad \quad \text{is\_subtype } ((x, t1) :: e) \text{ (freshen\_type } s2 x) \text{ (freshen\_type } t2 x)) \rightarrow \\ & \quad \text{is\_subtype } e \text{ (Forall } s1 s2) \text{ (Forall } t1 t2). \end{aligned}$$

The *sa\_all* rule for *Forall* types follows the pattern that we already introduced for the *wf\_type* predicate, in rule *wf\_type\_forall*. In the original, name-based specification, we say that  $E \vdash (\forall x <: \sigma_1. \sigma_2) <: (\forall x <: \tau_1. \tau_2)$  if  $E \vdash \tau_1 <: \sigma_1$  and  $E, x : \tau_1 \vdash \sigma_2 <: \tau_2$ . The type variable  $x$ , being  $\alpha$ -convertible in the conclusion, is (implicitly or explicitly) chosen so that  $E, x : \tau_1$  is well-formed in the second premise, that is,  $x$  is chosen not free in  $E$ .

In our locally nameless representation, the type variables bound by *Forall* in the conclusion do not have names. We must therefore invent a suitable name  $x$  and substitute it for the bound variable *TVar* 0 in the types  $s2$  and  $t2$ . Therefore, the second premise puts *freshen\_type*  $s2$   $x$  and *freshen\_type*  $t2$   $x$  in subtype relation.

As mentioned already,  $x$  should be chosen not in the domain of  $e$  (otherwise the extended environment  $(x, t1) :: e$  would be ill-formed) and not free in  $s2$  and  $t2$ , otherwise the freshenings *freshen\_type*  $s2$   $x$  and *freshen\_type*  $t2$   $x$  would incorrectly identify the bound, universally-quantified variable of the *Forall* types with an existing, free type variable. However, as we will prove below, the rules for *is\_subtype* satisfy a well-formedness condition: if *is\_subtype*  $e$   $u1$   $u2$ , then  $u1$  and  $u2$  are well-formed in  $e$ , implying that a name not in the domain of  $e$  cannot be free in  $u1$  or  $u2$ . Therefore, the condition “ $x$  not in the domain of  $e$ ” suffices to ensure that  $x$  is not free in  $s2$  and  $t2$ , and that the freshenings *freshen\_type*  $s2$   $x$  and *freshen\_type*  $t2$   $x$  make sense.

As mentioned already as well, we have a choice between quantifying over all suitable  $x$  or over one suitable  $x$  in the second premise. Again, we go with the “for all” alternative in order to obtain the strongest induction principle, and we will show later that the “for one” alternative is derivable.

For the time being, we start with simple well-formedness properties of the types and environments involved in a *is\_subtype* relation.

Lemma *is\_subtype\_wf\_env*:  $\forall e s t, \text{is\_subtype } e s t \rightarrow \text{wf\_env } e$ .

Lemma *is\_subtype\_wf\_type*:  $\forall e s t, \text{is\_subtype } e s t \rightarrow \text{wf\_type } e s \wedge \text{wf\_type } e t$ .

Lemma *is\_subtype\_wf\_type-l*:  $\forall e s t, \text{is\_subtype } e s t \rightarrow \text{wf\_type } e s$ .

Lemma *is\_subtype\_wf\_type-r*:  $\forall e s t, \text{is\_subtype } e s t \rightarrow \text{wf\_type } e t$ .

We now show that the *is\_subtype* predicate is stable by swapping. This property is crucial to show the equivalence of the “for all” and “for one” interpretations of rule *sa\_all*.

Lemma *is\_subtype\_swap*:

$\forall u v, \text{kind } u = \text{kind } v \rightarrow$

$\forall e s t, \text{is\_subtype } e s t \rightarrow$

$\text{is\_subtype } (\text{swap\_env } u v e) (\text{swap\_type } u v s) (\text{swap\_type } u v t)$ .

Two silly lemmas about freshness of names in types.

Lemma *fresh\_wf\_type*:  $\forall x e t, \text{wf\_type } e t \rightarrow \neg \text{In } x (\text{dom } e) \rightarrow \neg \text{In } x (\text{fv\_type } t)$ .

Lemma *fresh\_freshen\_type*:

$$\forall x \ t1 \ e \ t \ y, \\ wf\_type \ ((x, t1) :: e) \ (freshen\_type \ t \ x) \rightarrow \neg In \ y \ (dom \ e) \rightarrow x \neq y \rightarrow \neg In \ y \ (fv\_type \ t).$$

We now show that the alternate presentation of rule *sa\_all* (the one with “for one name” in the second premise instead of “for all names”) is admissible.

Lemma *sa\_all'*:

$$\forall e \ s1 \ s2 \ t1 \ t2 \ x, \\ is\_subtype \ e \ t1 \ s1 \rightarrow \\ kind \ x = TYPE \rightarrow \neg In \ x \ (dom \ e) \rightarrow \neg In \ x \ (fv\_type \ s2) \rightarrow \neg In \ x \ (fv\_type \ t2) \rightarrow \\ is\_subtype \ ((x, t1) :: e) \ (freshen\_type \ s2 \ x) \ (freshen\_type \ t2 \ x) \rightarrow \\ is\_subtype \ e \ (Forall \ s1 \ s2) \ (Forall \ t1 \ t2).$$

## 2.4 Reflexivity and transitivity of subtyping

We now turn (at last!) to proving the two theorems of part 1 of the POPLmark challenge: reflexivity and transitivity of subtyping.

Reflexivity of subtyping is shown by straightforward induction on the derivation of well-formedness of the type. As noted by McKinna and Pollack, such inductions conveniently replace inductions on the structure of types.

Theorem *sub\_refl*:  $\forall t \ e, wf\_type \ e \ t \rightarrow wf\_env \ e \rightarrow is\_subtype \ e \ t \ t.$

Proof.

*induction 1; intros.*

Case  $t = Tparam \ x$

*destruct (lookup\_exists \_ \_ H0) as [t L].*  
*apply sa\_refl\_tvar with t; auto.*

Case  $t = Top$

*apply sa\_top. auto. constructor.*

Case  $t = Arrow \ t1 \ t2$

*apply sa\_arrow. auto. auto.*

Case  $t = Forall \ t1 \ t2$

*destruct (fresh\_name TYPE (fv\_type t2 ++ dom e)) as [x [FRESH KIND]].*  
*apply sa\_all' with x; eauto.*  
*apply H1; eauto.*  
*constructor; eauto.*

Qed.

We now do some scaffolding work for the proof of transitivity. First, we will need to perform inductions over the size of types. We cannot just do inductions over the structure of types, as the original paper proof did, because in the case of *Forall t1 t2*, we will need to recurse

not on  $t2$  but on  $\text{freshen\_type } t2 \ x$  for some  $x$ , which is not a sub-term of  $t2$ . However, the size of  $\text{freshen\_type } t2 \ x$  is the same as the size of  $t2$ , so induction over sizes will work.

```

Fixpoint size_type (t: type): nat :=
  match t with
  | Tparam _ => 0
  | Tvar _ => 0
  | Top => 0
  | Arrow t1 t2 => 1 + size_type t1 + size_type t2
  | Forall t1 t2 => 1 + size_type t1 + size_type t2
  end.

```

Lemma  $\text{vsubst\_type\_size}$ :  $\forall x \ a \ n, \text{size\_type } (\text{vsubst\_type } a \ n \ (\text{Tparam } x)) = \text{size\_type } a$ .

Lemma  $\text{freshen\_type\_size}$ :  $\forall x \ a, \text{size\_type } (\text{freshen\_type } a \ x) = \text{size\_type } a$ .

We now define a notion of inclusion between environments that we call “weakening”.  $e2$  weakens  $e1$  if all the bindings in  $e1$  are preserved in  $e2$ ; however,  $e2$  may contain additional, non-interfering bindings.

```

Definition env_weaken (e1 e2: typenv) : Prop :=
   $\forall x \ t, \text{lookup } x \ e1 = \text{Some } t \rightarrow \text{lookup } x \ e2 = \text{Some } t$ .

```

Lemma  $\text{env\_weaken\_incl\_dom}$ :  $\forall e1 \ e2, \text{env\_weaken } e1 \ e2 \rightarrow \text{incl } (\text{dom } e1) \ (\text{dom } e2)$ .

The subtyping relation is stable by weakening of the typing environment.

Lemma  $\text{sub\_weaken}$ :

```

 $\forall e \ s \ t, \text{is\_subtype } e \ s \ t \rightarrow \forall e', \text{wf\_env } e' \rightarrow \text{env\_weaken } e \ e' \rightarrow \text{is\_subtype } e' \ s \ t$ .

```

A special case of weakening is the addition of bindings to an existing typing environment, provided the resulting environment is well-formed.

Lemma  $\text{env\_concat\_weaken}$ :

```

 $\forall \text{delta } \text{gamma},$ 
 $\text{wf\_env } (\text{delta} ++ \text{gamma}) \rightarrow \text{env\_weaken } \text{gamma} \ (\text{delta} ++ \text{gamma})$ .

```

The following lemmas prove useful properties of environments of the form  $e1 ++ (x, p) :: e2$ , that is, all bindings of  $e2$ , followed by a binding of  $p$  to  $x$ , followed by all bindings of  $e1$ .

Lemma  $\text{dom\_env\_extends}$ :  $\forall e1 \ x \ p \ q \ e2, \text{dom } (e1 ++ (x, p) :: e2) = \text{dom } (e1 ++ (x, q) :: e2)$ .

Lemma  $\text{wf\_env\_extends}$ :

```

 $\forall e2 \ x \ p \ q \ e1, \text{wf\_env } (e1 ++ (x, p) :: e2) \rightarrow \text{wf\_type } e2 \ q \rightarrow \text{wf\_env } (e1 ++ (x, q) :: e2)$ .

```

Lemma  $\text{lookup\_env\_extends}$ :

```

 $\forall e2 \ x \ p \ q \ y \ e1,$ 
 $\text{wf\_env } (e1 ++ (x, q) :: e2) \rightarrow$ 

```

*lookup*  $y$  ( $e1 ++ (x, p) :: e2$ ) = if *eq\_name*  $y$   $x$  then *Some*  $p$  else *lookup*  $y$  ( $e1 ++ (x, q) :: e2$ ).

Now comes the major result: transitivity and the narrowing property of subtyping, proved simultaneously. The proof follows the structure of the paper proof, with the structural induction on  $q$  being replaced by a Peano induction on the size of  $q$ .

Lemma *sub\_trans\_narrow*:

$$\begin{aligned} & \forall n, \\ & (\forall e\ s\ q\ t, \\ & \quad \text{size\_type } q \leq n \rightarrow \\ & \quad \text{is\_subtype } e\ s\ q \rightarrow \text{is\_subtype } e\ q\ t \rightarrow \\ & \quad \text{is\_subtype } e\ s\ t) \\ & \wedge (\forall x\ e1\ e2\ p\ q\ r\ s, \\ & \quad \text{size\_type } q \leq n \rightarrow \\ & \quad \text{is\_subtype } (e1 ++ (x, q) :: e2)\ r\ s \rightarrow \text{is\_subtype } e2\ p\ q \rightarrow \\ & \quad \text{is\_subtype } (e1 ++ (x, p) :: e2)\ r\ s). \end{aligned}$$

Proof.

*intro*  $n0$ . *pattern*  $n0$ . *apply* *Peano\_induction*.  
*intros* *size* *HRsize*.

Part 1: transitivity

*assert* ( $\forall e\ s\ q, \text{is\_subtype } e\ s\ q \rightarrow$   
 $\forall t, \text{size\_type } q \leq \text{size} \rightarrow \text{is\_subtype } e\ q\ t \rightarrow \text{is\_subtype } e\ s\ t$ ).

Sub-induction on the derivation of *is\_subtype*  $e\ s\ q$

*induction* 1; *intros*.

Case *sa\_top*

*inversion* *H2*. *apply* *sa\_top*. *auto*. *eauto*.

Case *sa\_refl\_tvar*

*auto*.

Case *sa\_trans\_tvar*

*apply* *sa\_trans\_tvar* with  $u$ ; *auto*.

Case *sa\_arrow*

*inversion* *H2*.

*apply* *sa\_top*. *auto*. *inversion* *H4*. *constructor*; *eauto*.

*subst*  $e0$ ; *subst*  $s0$ ; *subst*  $s3$ . *simpl* in *H1*.

*assert* (*SZpos*: *pred*  $\text{size} < \text{size}$ ). *omega*.

*elim* (*HRsize* (*pred*  $\text{size}$ ) *SZpos*); *intros* *HR1* *HR2*.

*apply* *sa\_arrow*.

Application of the outer induction hypothesis to  $t1$

*apply* *HR1* with  $t1$ ; *auto*. *omega*.

Application of the outer induction hypothesis to  $t2$

*apply* *HR1* with  $t2$ ; *auto*. *omega*.

Case *sa\_forall*

*inversion* *H3*.

*apply sa\_top. auto.*  
*apply is\_subtype\_wf\_type\_l with (Forall t1 t2). constructor; assumption.*  
*subst e0; subst s0; subst s3. simpl in H2.*  
*assert (SZpos: pred size < size). omega.*  
*elim (HRsize (pred size) SZpos); intros HR1 HR2.*  
 Choice of an appropriately fresh name  $x$   
*elim (fresh\_name TYPE (dom e ++ fv\_type t3 ++ fv\_type s2)).*  
*intros x [FRESH KIND].*  
*apply sa\_all' with x; eauto.*  
 Application of the outer induction hypothesis to  $t1$   
*apply HR1 with t1; auto. omega.*  
 Application of the outer induction hypothesis to *freshen t2 x*  
*apply HR1 with (freshen\_type t2 x).*  
*rewrite freshen\_type\_size. omega.*  
*change ((x, t0) :: e) with (nil ++ (x, t0) :: e).*  
 Application of the narrowing part of the outer induction hypothesis.  
*apply HR2 with t1. omega.*  
*simpl. apply H0; eauto. auto. apply H9; eauto.*  
 Part 2: narrowing  
*assert ( $\forall e r s,$   
*is\_subtype e r s  $\rightarrow$*   
 $\forall e1 x q e2 p,$   
 $e = e1 ++ (x, q) :: e2 \rightarrow \text{size\_type } q \leq \text{size} \rightarrow \text{is\_subtype } e2 p q \rightarrow$   
*is\_subtype (e1 ++ (x, p) :: e2) r s).*  
 Sub-induction on the derivation of *is\_subtype e r s*  
*induction 1; intros; subst e.*  
 Case *sa\_top*  
*apply sa\_top. apply wf\_env\_extends with q; eauto.*  
*apply wf\_type\_env\_incr with (e1 ++ (x, q) :: e2); auto.*  
*rewrite (dom\_env\_extends e1 x q p e2). apply incl\_refl.*  
 Case *sa\_refl\_tvar*  
*apply sa\_refl\_tvar with (if eq\_name x x0 then p else u).*  
*apply wf\_env\_extends with q; eauto. auto.*  
*rewrite (lookup\_env\_extends e2 x0 p q x e1 H0).*  
*case (eq\_name x x0); auto.*  
 Case *sa\_trans\_tvar*  
*apply sa\_trans\_tvar with (if eq\_name x x0 then p else u).*  
*auto. rewrite (lookup\_env\_extends e2 x0 p q x e1).*  
*case (eq\_name x x0); auto. eauto.*  
*case (eq\_name x x0); intro.*  
 sub-case  $x = x0$   
*generalize H1. rewrite (lookup\_env\_extends e2 x0 q q x e1); eauto.**

*rewrite e; rewrite eq\_name\_true; intro EQ; injection EQ; intro; subst u.*  
*apply H with q.*  
*apply sub\_weaken with e2; auto.*  
*apply wf\_env\_extends with q; eauto.*  
*change (e1 ++ (x0, p) :: e2) with (e1 ++ (((x0, p) :: nil) ++ e2)).*  
*rewrite ← app\_ass. apply env\_concat\_weaken.*  
*rewrite app\_ass. simpl. apply wf\_env\_extends with q; eauto.*  
*auto.*  
*apply IHis\_subtype with q; auto.*  
sub-case  $x \neq x0$   
*apply IHis\_subtype with q; auto.*  
Case *sa\_arrow*  
*apply sa\_arrow.*  
*apply IHis\_subtype1 with q; auto.*  
*apply IHis\_subtype2 with q; auto.*  
Case *sa\_forall*  
*apply sa\_all.*  
*apply IHis\_subtype with q; auto.*  
*intros.*  
*change ((x0, t1) :: e1 ++ (x, p) :: e2)*  
*with (((x0, t1) :: e1) ++ (x, p) :: e2).*  
*apply H2 with q. auto.*  
*rewrite (dom\_env\_extends e1 x q p e2). auto.*  
*reflexivity. auto. auto.*  
Combining the two parts together  
*split. intros; apply H with q; auto.*  
*intros; apply H0 with (e1 ++ (x, q) :: e2) q; auto.*  
Qed.

As a corollary, we obtain transitivity of subtyping.

Theorem *sub\_trans*:

$\forall e \ s \ q \ t, \text{is\_subtype } e \ s \ q \rightarrow \text{is\_subtype } e \ q \ t \rightarrow \text{is\_subtype } e \ s \ t.$

As well as narrowing.

Theorem *sub\_narrow*:

$\forall x \ e1 \ e2 \ p \ q \ r \ s,$   
 $\text{is\_subtype } (e1 \ ++ \ (x, q) \ :: \ e2) \ r \ s \rightarrow \text{is\_subtype } e2 \ p \ q \rightarrow$   
 $\text{is\_subtype } (e1 \ ++ \ (x, p) \ :: \ e2) \ r \ s.$

## 2.5 Stability of the subtyping judgement under substitutions

We now prove a property of the subtyping judgement that plays a crucial role later to prove that instantiation of polymorphic types is semantically sound. This property is stability under substitutions, namely that if  $\Gamma, X <: U \vdash T_1 <: T_2$ , then it must be the case that  $\Gamma \vdash T_1[X \leftarrow U] <: T_2[X \leftarrow U]$ . In the POPLmark challenge, this property belongs to part 2 of the challenge, but we prefer to prove it now, since it involves only the subtyping judgement.

To prove stability under substitution by induction, we need to strengthen its statement as follows: if  $\Gamma, X <: U, \Gamma' \vdash T_1 <: T_2$ , then  $\Gamma, \Gamma'[X \leftarrow U] \vdash T_1[X \leftarrow U] <: T_2[X \leftarrow U]$ . We therefore need to extend type substitutions to environments, pointwise.

```
Fixpoint psubst_env (e: typenv) (x: name) (b: type) {struct e} : typenv :=
  match e with
  | nil => nil
  | (y, t) :: e' => (y, psubst_type t x b) :: psubst_env e' x b
  end.
```

Lemma *dom\_psubst\_env*:  $\forall x b e, \text{dom} (\text{psubst\_env } e x b) = \text{dom } e$ .

### 2.5.1 Commutation properties for type substitutions

As a preliminary result, we need to show that well-formedness of types and environments is preserved by substitution of a well-formed type for a name. This simple property needs a number of commutation properties between the *psubst\_type*, *freshen\_type* and *vsubst\_type* functions. Unfortunately, some hacking on de Bruijn indices is necessary. We start by defining a predicate *type\_vars\_below t n* that holds if all free de Bruijn variables in the type *t* are less than *n*. In particular, *type\_vars\_below t 0* holds iff *t* has no free de Bruijn variables.

```
Fixpoint type_vars_below (t: type) (n: nat) {struct t} : Prop :=
  match t with
  | Tparam x => True
  | Tvar m => m < n
  | Top => True
  | Arrow t1 t2 => type_vars_below t1 n & type_vars_below t2 n
  | Forall t1 t2 => type_vars_below t1 n & type_vars_below t2 (S n)
  end.
```

Lemma *type\_vars\_below\_vsubst*:  
 $\forall t n t', \text{type\_vars\_below} (\text{vsubst\_type } t n t') n \rightarrow \text{type\_vars\_below } t (S n)$ .



A well-formed type, having no free de Bruijn variables, has all its de Bruijn variables below 0.

Lemma *wf\_type\_vars\_below\_0*:  $\forall e t, wf\_type\ e\ t \rightarrow type\_vars\_below\ t\ 0$ .

A type is invariant by *vsubst\_type* substitution if the de Bruijn variable being substituted is not free in the type.

Lemma *vsubst\_invariant\_below*:

$$\forall t\ n\ m\ s, type\_vars\_below\ t\ n \rightarrow n \leq m \rightarrow vsubst\_type\ t\ m\ s = t.$$

As a corollary, well-formed types are invariant by *vsubst* substitutions.

Lemma *vsubst\_wf\_type*:  $\forall e\ t\ n\ s, wf\_type\ e\ t \rightarrow vsubst\_type\ t\ n\ s = t$ .

It follows that *vsubst* and *psubst* substitutions commute in the following sense.

Lemma *psubst\_vsubst\_type*:

$$\begin{aligned} &\forall e\ a\ x\ b\ n\ c, \\ &wf\_type\ e\ b \rightarrow \\ &vsubst\_type\ (psubst\_type\ a\ x\ b)\ n\ (psubst\_type\ c\ x\ b) = \\ &psubst\_type\ (vsubst\_type\ a\ n\ c)\ x\ b. \end{aligned}$$

Consequently, *psubst\_type* and *freshen\_type* commute if they operate on distinct names.

Lemma *psubst\_freshen\_type*:

$$\begin{aligned} &\forall e\ a\ x\ b\ y, \\ &wf\_type\ e\ b \rightarrow x \neq y \rightarrow \\ &freshen\_type\ (psubst\_type\ a\ x\ b)\ y = psubst\_type\ (freshen\_type\ a\ y)\ x\ b. \end{aligned}$$

Additionally, *vsubst* substitution of de Bruijn variable 0 is equivalent to freshening with a fresh type name *x* followed by a *psubst* substitution over *x*.

Lemma *vsubst\_psubst\_type*:

$$\begin{aligned} &\forall x\ t2\ t1\ n, \neg In\ x\ (fv\_type\ t1) \rightarrow \\ &vsubst\_type\ t1\ n\ t2 = psubst\_type\ (vsubst\_type\ t1\ n\ (Tparam\ x))\ x\ t2. \end{aligned}$$

Lemma *vsubst\_psubst\_freshen\_type*:

$$\begin{aligned} &\forall x\ t1\ t2, \neg In\ x\ (fv\_type\ t1) \rightarrow \\ &vsubst\_type\ t1\ 0\ t2 = psubst\_type\ (freshen\_type\ t1\ x)\ x\ t2. \end{aligned}$$

## 2.5.2 Preservation of well-formedness of types and environments during substitution

Well-formedness of types is preserved by *psubst\_type* substitution of a well-formed type for a type name.

Lemma *wf\_type\_psubst*:

$$\begin{aligned} & \forall e1\ x\ p\ q\ e2\ t, \\ & wf\_type\ (e2\ ++\ (x,\ p)\ ::\ e1)\ t \rightarrow wf\_type\ e1\ q \rightarrow \\ & wf\_type\ (psubst\_env\ e2\ x\ q\ ++\ e1)\ (psubst\_type\ t\ x\ q). \end{aligned}$$

Similarly, well-formedness of environments is preserved by *psubst\_env* substitution.

Lemma *wf\_env\_psubst*:

$$\begin{aligned} & \forall e1\ x\ p\ q\ e2, \\ & wf\_env\ (e2\ ++\ (x,\ p)\ ::\ e1) \rightarrow wf\_type\ e1\ q \rightarrow \\ & wf\_env\ (psubst\_env\ e2\ x\ q\ ++\ e1). \end{aligned}$$

### 2.5.3 Type substitution preserves subtyping

We now show that if *s* is subtype of *t* in the environment  $e2\ ++\ (x,\ p)\ ::\ e1$ , and if *q* is subtype of *p* in *e1*, then *psubst s x q* is subtype of *psubst t x q* in *psubst\_env x q e2 ++ e1*.

Well-formed environments of the form  $e2\ ++\ (x,\ p)\ ::\ e1$  are such that *x* is not in the domain of *e1* and not free in the types listed in *e1*. Therefore, these types are invariant by substitution over *x*.

Lemma *env\_concat\_not\_free*:  $\forall e1\ x\ p\ e2, wf\_env\ (e2\ ++\ (x,\ p)\ ::\ e1) \rightarrow \neg In\ x\ (dom\ e1)$ .

Lemma *psubst\_type\_inv*:  $\forall t\ x\ s, \neg In\ x\ (fv\_type\ t) \rightarrow psubst\_type\ t\ x\ s = t$ .

The following technical lemma relates *lookup* operations between the original environment  $e2\ ++\ (x,\ p)\ ::\ e1$  and its substituted counterpart, *psubst\_env e2 x q ++ e1*.

Lemma *lookup\_env\_concat*:

$$\begin{aligned} & \forall e1\ x\ p\ q\ y\ t\ e2, \\ & wf\_env\ (e2\ ++\ (x,\ p)\ ::\ e1) \rightarrow \\ & lookup\ y\ (e2\ ++\ (x,\ p)\ ::\ e1) = Some\ t \rightarrow \\ & (x = y \wedge t = p) \vee \\ & (x \neq y \wedge lookup\ y\ (psubst\_env\ e2\ x\ q\ ++\ e1) = Some\ (psubst\_type\ t\ x\ q)). \end{aligned}$$

We can now prove the stability of subtyping by type substitution (lemma A.10 in the challenge statement). The proof proceeds by induction over the derivation of the second subtyping hypothesis.

Theorem *sub\_stable\_subst*:

$$\begin{aligned} & \forall e1\ x\ p\ q\ e2\ s\ t, \\ & is\_subtype\ e1\ p\ q \rightarrow \\ & is\_subtype\ (e2\ ++\ (x,\ q)\ ::\ e1)\ s\ t \rightarrow \\ & is\_subtype\ (psubst\_env\ e2\ x\ p\ ++\ e1)\ (psubst\_type\ s\ x\ p)\ (psubst\_type\ t\ x\ p). \end{aligned}$$

Proof.

$$\begin{aligned} & assert\ (\forall e1\ x\ p\ q, is\_subtype\ e1\ p\ q \rightarrow \\ & \quad \forall e\ s\ t, is\_subtype\ e\ s\ t \rightarrow \end{aligned}$$

$\forall e2, e = e2 ++ (x, q) :: e1 \rightarrow$   
 $is\_subtype (psubst\_env e2 x p ++ e1)$   
 $(psubst\_type s x p) (psubst\_type t x p)).$

*induction 2; intros; simpl; subst e.*

Rule *sa\_top*  
*constructor. eapply wf\_env\_psubst; eauto.*  
*eapply wf\_type\_psubst; eauto.*

Rule *sa\_refl*  
*apply sub\_refl.*  
*case (eq\_name x0 x); intro.*

Rule *sa\_refl*, case  $x = x0$   
*apply wf\_type\_env\_incr with e1; eauto.*  
*rewrite dom\_append. apply incl\_appr. apply incl\_refl.*

Rule *sa\_refl*, case  $x \neq x0$   
*constructor. auto. generalize (lookup\_inv \_ \_ \_ H2).*  
*repeat rewrite dom\_append. simpl. rewrite dom\_psubst\_env.*  
*intro. apply in\_or\_app. elim (in\_app\_or \_ \_ \_ H3); intro.*  
*auto. elim H4; intro. congruence. auto.*  
*eapply wf\_env\_psubst; eauto.*

Rule *sa\_trans*  
*assert (wf\_env (e2 ++ (x, q) :: e1)). eapply is\_subtype\_wf\_env; eauto.*  
*generalize (lookup\_env\_concat \_ \_ \_ p \_ \_ \_ H3 H1). intro.*  
*case (eq\_name x0 x); intro.*

Rule *sa\_trans*, case  $x = x0$   
*subst x0. assert (u = q). intuition. subst u.*  
*apply sub\_trans with (psubst\_type q x p).*  
*replace (psubst\_type q x p) with q.*  
*apply sub\_weaken with e1. auto. eapply wf\_env\_psubst; eauto.*  
*apply env\_concat\_weaken. eapply wf\_env\_psubst; eauto.*  
*symmetry. apply psubst\_type\_inv.*  
*assert (wf\_type e1 q). eapply is\_subtype\_wf\_type\_r; eauto.*  
*generalize (fv\_wf\_type x \_ \_ H5).*  
*generalize (env\_concat\_not\_free \_ \_ \_ \_ H3). tauto.*  
*apply IHis\_subtype. auto.*

Rule *sa\_trans*, case  $x \neq x0$   
*apply sa\_trans\_tvar with (psubst\_type u x p).*  
*auto. generalize (sym\_not\_equal n); tauto.*  
*apply IHis\_subtype. auto.*

Rule *sa\_arrow*  
*constructor; auto.*

Rule *sa\_forall*  
*destruct (fresh\_name TYPE (x :: dom (e2 ++ (x, q) :: e1) ++*

```

                                dom (psubst_env e2 x p ++ e1) ++
                                fv_type (psubst_type s2 x p) ++
                                fv_type (psubst_type t2 x p))
as [y [F K]].
eapply sa_all'; eauto.
repeat rewrite (psubst_freshen_type e1).
change ((y, psubst_type t1 x p) :: psubst_env e2 x p ++ e1)
  with (psubst_env ((y, t1) :: e2) x p ++ e1).
apply H2; eauto. eauto. eauto. eauto. eauto.
eauto.
Qed.
```



## Chapter 3

# Type soundness

This chapter addresses part 2A of the POPLmark challenge, namely the proof of soundness of the  $F_{<}$  type systems without records.

```
Require Import Arith.  
Require Import ZArith.  
Require Import List.  
Require Import extralibrary.  
Require Import part1a.
```

### 3.1 Terms

We now define the syntax of  $F_{<}$  terms and basic syntactic notions such as free variables, substitutions, and well-formedness of terms. We follow the same approach used for types in chapter 2.

#### 3.1.1 Syntax and syntactic operations

The syntax of terms is defined as follows. As in types, bound variables are represented by de Bruijn indices, while free variables are represented by names. Bound term variables and bound type variables are numbered independently. In a lambda-abstraction  $TFun\ t\ a$ , the term variable  $Var\ 0$  is bound in  $a$ . In a type abstraction  $TApp\ t\ a$ , the type variable  $TVar\ 0$  is bound in  $a$ .

```
Inductive term: Set :=  
  | Param: name → term  
  | Var: nat → term  
  | Fun: type → term → term
```

```

| App: term → term → term
| TFun: type → term → term
| TApp: term → type → term.

```

The free names of a term include both type names and term names.

```

Fixpoint fv_term (a: term) : list name :=
  match a with
  | Param v ⇒ v :: nil
  | Var n ⇒ nil
  | Fun t a1 ⇒ fv_type t ++ fv_term a1
  | App a1 a2 ⇒ fv_term a1 ++ fv_term a2
  | TFun t a1 ⇒ fv_type t ++ fv_term a1
  | TApp a1 t ⇒ fv_term a1 ++ fv_type t
  end.

```

There are 4 substitution operations over terms, depending on whether we are substituting a named variable (*psubst\_*) or a de Bruijn variable (*vsubst\_*), and whether we are substituting a term for a term variable (*\_term*) or a type for a type variable (*\_tety*).

```

Fixpoint vsubst_term (a: term) (x: nat) (b: term) {struct a} : term :=
  match a with
  | Param v ⇒ Param v
  | Var n ⇒
      match compare_nat n x with
      | Nat_less _ ⇒ Var n
      | Nat_equal _ ⇒ b
      | Nat_greater _ ⇒ Var (pred n)
      end
  | Fun t a1 ⇒ Fun t (vsubst_term a1 (S x) b)
  | App a1 a2 ⇒ App (vsubst_term a1 x b) (vsubst_term a2 x b)
  | TFun t a1 ⇒ TFun t (vsubst_term a1 x b)
  | TApp a1 t ⇒ TApp (vsubst_term a1 x b) t
  end.

```

```

Fixpoint psubst_term (a: term) (x: name) (b: term) {struct a} : term :=
  match a with
  | Param v ⇒ if eq_name v x then b else Param v
  | Var n ⇒ Var n
  | Fun t a1 ⇒ Fun t (psubst_term a1 x b)
  | App a1 a2 ⇒ App (psubst_term a1 x b) (psubst_term a2 x b)
  | TFun t a1 ⇒ TFun t (psubst_term a1 x b)
  | TApp a1 t ⇒ TApp (psubst_term a1 x b) t
  end.

```

```

Fixpoint vsubst_tety (a: term) (x: nat) (b: type) {struct a} : term :=

```

```

match a with
| Param v ⇒ Param v
| Var n ⇒ Var n
| Fun t a1 ⇒ Fun (vsubst_type t x b) (vsubst_tety a1 x b)
| App a1 a2 ⇒ App (vsubst_tety a1 x b) (vsubst_tety a2 x b)
| TFun t a1 ⇒ TFun (vsubst_type t x b) (vsubst_tety a1 (S x) b)
| TApp a1 t ⇒ TApp (vsubst_tety a1 x b) (vsubst_type t x b)
end.

```

```

Fixpoint psubst_tety (a: term) (x: name) (b: type) {struct a} : term :=
  match a with
  | Param v ⇒ Param v
  | Var n ⇒ Var n
  | Fun t a1 ⇒ Fun (psubst_type t x b) (psubst_tety a1 x b)
  | App a1 a2 ⇒ App (psubst_tety a1 x b) (psubst_tety a2 x b)
  | TFun t a1 ⇒ TFun (psubst_type t x b) (psubst_tety a1 x b)
  | TApp a1 t ⇒ TApp (psubst_tety a1 x b) (psubst_type t x b)
  end.

```

Here are the two “freshening” operations that replace the bound variable 0 with a term or type name, respectively.

```

Definition freshen_term (a: term) (x: name) : term :=
  vsubst_term a 0 (Param x).

```

```

Definition freshen_tety (a: term) (x: name) : term :=
  vsubst_tety a 0 (Tparam x).

```

Substitutions and freshening play well with free variables.

```

Lemma fv_term_vsubst_term:
  ∀ x a n b, In x (fv_term a) → In x (fv_term (vsubst_term a n b)).

```

```

Lemma fv_term_vsubst_tety:
  ∀ x a n b, In x (fv_term a) → In x (fv_term (vsubst_tety a n b)).

```

```

Lemma fv_term_freshen_term:
  ∀ x a y, In x (fv_term a) → In x (fv_term (freshen_term a y)).

```

```

Lemma fv_term_freshen_tety:
  ∀ x a y, In x (fv_term a) → In x (fv_term (freshen_tety a y)).

```

Swaps of two names in a term.

```

Fixpoint swap_term (u v: name) (a: term) {struct a} : term :=
  match a with
  | Param x ⇒ Param (swap u v x)
  | Var n ⇒ Var n

```



|  $Fun\ t\ a1 \Rightarrow Fun\ (swap\_type\ u\ v\ t)\ (swap\_term\ u\ v\ a1)$   
|  $App\ a1\ a2 \Rightarrow App\ (swap\_term\ u\ v\ a1)\ (swap\_term\ u\ v\ a2)$   
|  $TFun\ t\ a1 \Rightarrow TFun\ (swap\_type\ u\ v\ t)\ (swap\_term\ u\ v\ a1)$   
|  $TApp\ a1\ t \Rightarrow TApp\ (swap\_term\ u\ v\ a1)\ (swap\_type\ u\ v\ t)$   
*end.*

Swaps commute with the free variables operation.

Lemma *in\_fv\_term\_swap*:

$\forall u\ v\ x\ a,$   
 $In\ x\ (fv\_term\ a) \leftrightarrow In\ (swap\ u\ v\ x)\ (fv\_term\ (swap\_term\ u\ v\ a)).$

Lemma *swap\_term\_not\_free*:

$\forall u\ v\ a, \neg In\ u\ (fv\_term\ a) \rightarrow \neg In\ v\ (fv\_term\ a) \rightarrow swap\_term\ u\ v\ a = a.$

Swaps are self-inverse.

Lemma *swap\_term\_inv*:  $\forall u\ v\ a, swap\_term\ u\ v\ (swap\_term\ u\ v\ a) = a.$

Swaps commute with substitutions and freshening.

Lemma *vsubst\_term\_swap*:

$\forall u\ v\ a\ n\ b,$   
 $swap\_term\ u\ v\ (vsubst\_term\ a\ n\ b) =$   
 $vsubst\_term\ (swap\_term\ u\ v\ a)\ n\ (swap\_term\ u\ v\ b).$

Lemma *vsubst\_tety\_swap*:

$\forall u\ v\ a\ n\ b,$   
 $swap\_term\ u\ v\ (vsubst\_tety\ a\ n\ b) =$   
 $vsubst\_tety\ (swap\_term\ u\ v\ a)\ n\ (swap\_type\ u\ v\ b).$

Lemma *freshen\_term\_swap*:

$\forall u\ v\ a\ x,$   
 $swap\_term\ u\ v\ (freshen\_term\ a\ x) =$   
 $freshen\_term\ (swap\_term\ u\ v\ a)\ (swap\ u\ v\ x).$

Lemma *freshen\_tety\_swap*:

$\forall u\ v\ a\ x,$   
 $swap\_term\ u\ v\ (freshen\_tety\ a\ x) =$   
 $freshen\_tety\ (swap\_term\ u\ v\ a)\ (swap\ u\ v\ x).$

### 3.1.2 Well-formedness of terms

A term is well-formed in a typing environment if:

- all types contained within are well-formed as per *wf\_type*;

- all names  $n$  appearing free in a  $Param\ n$  subterm are of kind  $TERM$  and are bound in the environment;
- it does not contain free de Bruijn variables.

Inductive  $wf\_term: typenv \rightarrow term \rightarrow Prop :=$

- |  $wf\_term\_param: \forall e\ x,$   
 $kind\ x = TERM \rightarrow In\ x\ (dom\ e) \rightarrow$   
 $wf\_term\ e\ (Param\ x)$
- |  $wf\_term\_fun: \forall e\ t\ a,$   
 $wf\_type\ e\ t \rightarrow$   
 $(\forall x,$   
 $kind\ x = TERM \rightarrow$   
 $\neg In\ x\ (fv\_term\ a) \rightarrow \neg In\ x\ (dom\ e) \rightarrow$   
 $wf\_term\ ((x, t) :: e)\ (freshen\_term\ a\ x)) \rightarrow$   
 $wf\_term\ e\ (Fun\ t\ a)$
- |  $wf\_term\_app: \forall e\ a1\ a2,$   
 $wf\_term\ e\ a1 \rightarrow wf\_term\ e\ a2 \rightarrow$   
 $wf\_term\ e\ (App\ a1\ a2)$
- |  $wf\_term\_tfun: \forall e\ t\ a,$   
 $wf\_type\ e\ t \rightarrow$   
 $(\forall x,$   
 $kind\ x = TYPE \rightarrow$   
 $\neg In\ x\ (fv\_term\ a) \rightarrow \neg In\ x\ (dom\ e) \rightarrow$   
 $wf\_term\ ((x, t) :: e)\ (freshen\_tety\ a\ x)) \rightarrow$   
 $wf\_term\ e\ (TFun\ t\ a)$
- |  $wf\_term\_tapp: \forall e\ a\ t,$   
 $wf\_term\ e\ a \rightarrow wf\_type\ e\ t \rightarrow$   
 $wf\_term\ e\ (TApp\ a\ t).$

A term well formed in  $e$  has all its free names in the domain of  $e$ .

Lemma  $fv\_wf\_term: \forall x\ e\ t, wf\_term\ e\ t \rightarrow In\ x\ (fv\_term\ t) \rightarrow In\ x\ (dom\ e)$ .

Well-formedness is stable under swaps.

Lemma  $wf\_term\_swap:$

- $\forall u\ v, kind\ u = kind\ v \rightarrow$
- $\forall e\ a, wf\_term\ e\ a \rightarrow wf\_term\ (swap\_env\ u\ v\ e)\ (swap\_term\ u\ v\ a).$

A term well-formed in  $e$  remains well-formed if extra bindings are added to  $e$ .

Lemma  $wf\_term\_env\_incr:$

- $\forall e\ a, wf\_term\ e\ a \rightarrow \forall e', incl\ (dom\ e)\ (dom\ e') \rightarrow wf\_term\ e'\ a.$

Here are two admissible rules that prove the well-formedness of  $Fun$  and  $TFun$  abstractions. These rules are similar to the  $wf\_term\_fun$  and  $wf\_term\_tfun$  rules, but with a premise of the form “there exists a name” instead of the original “for all names”.

Lemma *wf\_term\_fun'*:

$$\begin{aligned} & \forall e x t a, \\ & \text{wf\_type } e t \rightarrow \\ & \text{kind } x = \text{TERM} \rightarrow \neg \text{In } x (\text{fv\_term } a) \rightarrow \neg \text{In } x (\text{dom } e) \rightarrow \\ & \text{wf\_term } ((x, t) :: e) (\text{freshen\_term } a x) \rightarrow \\ & \text{wf\_term } e (\text{Fun } t a). \end{aligned}$$

Lemma *wf\_term\_tfun'*:

$$\begin{aligned} & \forall e x t a, \\ & \text{wf\_type } e t \rightarrow \\ & \text{kind } x = \text{TYPE} \rightarrow \neg \text{In } x (\text{fv\_term } a) \rightarrow \neg \text{In } x (\text{dom } e) \rightarrow \\ & \text{wf\_term } ((x, t) :: e) (\text{freshen\_tety } a x) \rightarrow \\ & \text{wf\_term } e (\text{TFun } t a). \end{aligned}$$

### 3.1.3 Properties of term substitutions

To prove the usual properties of term substitutions, we follow the same approach as for type substitutions, starting with a characterization of terms that have no free de Bruijn variables, or all such variables below some threshold.

Fixpoint *term\_vars\_below* (*a*: term) (*nterm ntype*: nat) {*struct a*} : Prop :=

$$\begin{aligned} & \text{match } a \text{ with} \\ & | \text{Param } x \Rightarrow \text{True} \\ & | \text{Var } n \Rightarrow n < \text{nterm} \\ & | \text{Fun } t b \Rightarrow \text{type\_vars\_below } t \text{ ntype} \wedge \text{term\_vars\_below } b (S \text{ nterm}) \text{ ntype} \\ & | \text{App } b c \Rightarrow \text{term\_vars\_below } b \text{ nterm ntype} \wedge \text{term\_vars\_below } c \text{ nterm ntype} \\ & | \text{TFun } t b \Rightarrow \text{type\_vars\_below } t \text{ ntype} \wedge \text{term\_vars\_below } b \text{ nterm } (S \text{ ntype}) \\ & | \text{TApp } b t \Rightarrow \text{term\_vars\_below } b \text{ nterm ntype} \wedge \text{type\_vars\_below } t \text{ ntype} \\ & \text{end.} \end{aligned}$$

Lemma *term\_vars\_below\_vsubst\_term*:

$$\begin{aligned} & \forall a \text{ nterm ntype } a', \\ & \text{term\_vars\_below } (\text{vsubst\_term } a \text{ nterm } a') \text{ nterm ntype} \rightarrow \\ & \text{term\_vars\_below } a (S \text{ nterm}) \text{ ntype}. \end{aligned}$$

Lemma *term\_vars\_below\_vsubst\_tety*:

$$\begin{aligned} & \forall a \text{ nterm ntype } a', \\ & \text{term\_vars\_below } (\text{vsubst\_tety } a \text{ ntype } a') \text{ nterm ntype} \rightarrow \\ & \text{term\_vars\_below } a \text{ nterm } (S \text{ ntype}). \end{aligned}$$

Lemma *wf\_term\_vars\_below\_0*:  $\forall e a, \text{wf\_term } e a \rightarrow \text{term\_vars\_below } a \ 0 \ 0$ .

Lemma *vsubst\_term\_invariant\_below*:

$$\forall a \text{ n1 n2 } m b, \text{term\_vars\_below } a \text{ n1 } n2 \rightarrow n1 \leq m \rightarrow \text{vsubst\_term } a \ m \ b = a.$$

Lemma *vsubst\_tety\_invariant\_below*:

$$\forall a \ n1 \ n2 \ m \ t, \text{term\_vars\_below } a \ n1 \ n2 \rightarrow n2 \leq m \rightarrow \text{vsubst\_tety } a \ m \ t = a.$$

Lemma *vsubst\_term\_wf\_term*:

$$\forall e \ a \ n \ b, \text{wf\_term } e \ a \rightarrow \text{vsubst\_term } a \ n \ b = a.$$

Lemma *vsubst\_tety\_wf\_term*:

$$\forall e \ a \ n \ t, \text{wf\_term } e \ a \rightarrow \text{vsubst\_tety } a \ n \ t = a.$$

Lemma *psubst\_vsubst\_term*:

$$\begin{aligned} &\forall e \ a \ x \ b \ n \ c, \\ &\text{wf\_term } e \ b \rightarrow \\ &\text{vsubst\_term } (\text{psubst\_term } a \ x \ b) \ n \ (\text{psubst\_term } c \ x \ b) = \\ &\text{psubst\_term } (\text{vsubst\_term } a \ n \ c) \ x \ b. \end{aligned}$$

Lemma *psubst\_freshen\_term*:

$$\begin{aligned} &\forall e \ a \ x \ b \ y, \\ &\text{wf\_term } e \ b \rightarrow x \neq y \rightarrow \\ &\text{freshen\_term } (\text{psubst\_term } a \ x \ b) \ y = \text{psubst\_term } (\text{freshen\_term } a \ y) \ x \ b. \end{aligned}$$

Lemma *psubst\_vsubst\_tety*:

$$\begin{aligned} &\forall e \ a \ x \ b \ n \ c, \\ &\text{wf\_type } e \ b \rightarrow \\ &\text{vsubst\_tety } (\text{psubst\_tety } a \ x \ b) \ n \ (\text{psubst\_type } c \ x \ b) = \\ &\text{psubst\_tety } (\text{vsubst\_tety } a \ n \ c) \ x \ b. \end{aligned}$$

Lemma *psubst\_freshen\_tety*:

$$\begin{aligned} &\forall e \ a \ x \ b \ y, \\ &\text{wf\_type } e \ b \rightarrow x \neq y \rightarrow \\ &\text{freshen\_tety } (\text{psubst\_tety } a \ x \ b) \ y = \text{psubst\_tety } (\text{freshen\_tety } a \ y) \ x \ b. \end{aligned}$$

Lemma *psubst\_vsubst\_tetyte*:

$$\begin{aligned} &\forall e \ a \ x \ b \ n \ c, \\ &\text{wf\_type } e \ b \rightarrow \\ &\text{vsubst\_term } (\text{psubst\_tety } a \ x \ b) \ n \ (\text{psubst\_tety } c \ x \ b) = \\ &\text{psubst\_tety } (\text{vsubst\_term } a \ n \ c) \ x \ b. \end{aligned}$$

Lemma *psubst\_freshen\_tetyte*:

$$\begin{aligned} &\forall e \ a \ x \ b \ y, \\ &\text{wf\_type } e \ b \rightarrow x \neq y \rightarrow \\ &\text{freshen\_term } (\text{psubst\_tety } a \ x \ b) \ y = \text{psubst\_tety } (\text{freshen\_term } a \ y) \ x \ b. \end{aligned}$$

Lemma *psubst\_vsubst\_tetyte*:

$$\begin{aligned} &\forall e \ a \ x \ b \ n \ c, \\ &\text{wf\_term } e \ b \rightarrow \\ &\text{vsubst\_tety } (\text{psubst\_term } a \ x \ b) \ n \ c = \text{psubst\_term } (\text{vsubst\_tety } a \ n \ c) \ x \ b. \end{aligned}$$

Lemma *psubst\_freshen\_tetyte*:

$$\begin{aligned} &\forall e a x b y, \\ &wf\_term e b \rightarrow x \neq y \rightarrow \\ &freshen\_tety (psubst\_term a x b) y = psubst\_term (freshen\_tety a y) x b. \end{aligned}$$

Lemma *vsubst\_psubst\_term*:

$$\begin{aligned} &\forall x a2 a1 n, \\ &\neg In x (fv\_term a1) \rightarrow \\ &vsubst\_term a1 n a2 = psubst\_term (vsubst\_term a1 n (Param x)) x a2. \end{aligned}$$

Lemma *vsubst\_psubst\_freshen\_term*:

$$\begin{aligned} &\forall x a1 a2, \\ &\neg In x (fv\_term a1) \rightarrow \\ &vsubst\_term a1 0 a2 = psubst\_term (freshen\_term a1 x) x a2. \end{aligned}$$

Lemma *vsubst\_psubst\_tety*:

$$\begin{aligned} &\forall x t2 a1 n, \\ &\neg In x (fv\_term a1) \rightarrow \\ &vsubst\_tety a1 n t2 = psubst\_tety (vsubst\_tety a1 n (Tparam x)) x t2. \end{aligned}$$

Lemma *vsubst\_psubst\_freshen\_tety*:

$$\begin{aligned} &\forall x a1 t2, \\ &\neg In x (fv\_term a1) \rightarrow \\ &vsubst\_tety a1 0 t2 = psubst\_tety (freshen\_tety a1 x) x t2. \end{aligned}$$

## 3.2 Typing rules

We now define the typing judgement “term  $a$  has type  $t$  in environment  $e$ ” as an inductive predicate *has\_type e a t*.

Inductive *has\_type*: *typenv*  $\rightarrow$  *term*  $\rightarrow$  *type*  $\rightarrow$  *Prop* :=

$$\begin{aligned} &| t\_var: \forall e x t, \\ &\quad wf\_env e \rightarrow kind x = TERM \rightarrow lookup x e = Some t \rightarrow \\ &\quad has\_type e (Param x) t \\ &| t\_abs: \forall e t1 a t2, \\ &\quad wf\_type e t1 \rightarrow \\ &\quad (\forall x, \\ &\quad\quad kind x = TERM \rightarrow \neg In x (dom e) \rightarrow \\ &\quad\quad has\_type ((x, t1) :: e) (freshen\_term a x) t2) \rightarrow \\ &\quad has\_type e (Fun t1 a) (Arrow t1 t2) \\ &| t\_app: \forall e a b t1 t2, \\ &\quad has\_type e a (Arrow t1 t2) \rightarrow has\_type e b t1 \rightarrow \\ &\quad has\_type e (App a b) t2 \\ &| t\_tabs: \forall e t1 a t2, \\ &\quad wf\_type e t1 \rightarrow \end{aligned}$$

$$\begin{aligned}
& (\forall x, \\
& \quad \text{kind } x = \text{TYPE} \rightarrow \neg \text{In } x (\text{dom } e) \rightarrow \\
& \quad \text{has\_type } ((x, t1) :: e) (\text{freshen\_tety } a \ x) (\text{freshen\_type } t2 \ x) \rightarrow \\
& \quad \text{has\_type } e (\text{TFun } t1 \ a) (\text{Forall } t1 \ t2) \\
| \ t\_tapp: & \forall e \ a \ t \ t1 \ t2, \\
& \quad \text{has\_type } e \ a (\text{Forall } t1 \ t2) \rightarrow \\
& \quad \text{is\_subtype } e \ t \ t1 \rightarrow \\
& \quad \text{has\_type } e (\text{TApp } a \ t) (\text{vsubst\_type } t2 \ O \ t) \\
| \ t\_sub: & \forall e \ a \ t1 \ t2, \\
& \quad \text{has\_type } e \ a \ t1 \rightarrow \text{is\_subtype } e \ t1 \ t2 \rightarrow \\
& \quad \text{has\_type } e \ a \ t2.
\end{aligned}$$

Well-formedness properties: if  $\text{has\_type } e \ a \ t$  holds, then  $e$  is a well-formed environment,  $t$  a well-formed type and  $a$  a well-formed term.

Lemma  $\text{has\_type\_wf\_env}$ :  $\forall e \ a \ t, \text{has\_type } e \ a \ t \rightarrow \text{wf\_env } e$ .

Lemma  $\text{wf\_type\_strengthen}$ :

$$\begin{aligned}
& \forall e \ t, \text{wf\_type } e \ t \rightarrow \\
& \forall e', (\forall x, \text{kind } x = \text{TYPE} \rightarrow \text{In } x (\text{dom } e) \rightarrow \text{In } x (\text{dom } e')) \rightarrow \text{wf\_type } e' \ t.
\end{aligned}$$

Lemma  $\text{has\_type\_wf\_type}$ :  $\forall e \ a \ t, \text{has\_type } e \ a \ t \rightarrow \text{wf\_type } e \ t$ .

Lemma  $\text{has\_type\_wf\_term}$ :  $\forall e \ a \ t, \text{has\_type } e \ a \ t \rightarrow \text{wf\_term } e \ a$ .

The  $\text{has\_type}$  predicate is stable by addition of hypotheses.

Lemma  $\text{wf\_type\_weaken}$ :  $\forall e \ e' \ t, \text{wf\_type } e \ t \rightarrow \text{env\_weaken } e \ e' \rightarrow \text{wf\_type } e' \ t$ .

Lemma  $\text{wf\_term\_weaken}$ :  $\forall e \ e' \ a, \text{wf\_term } e \ a \rightarrow \text{env\_weaken } e \ e' \rightarrow \text{wf\_term } e' \ a$ .

Lemma  $\text{env\_weaken\_add}$ :

$$\forall e \ e' \ x \ t, \text{env\_weaken } e \ e' \rightarrow \text{env\_weaken } ((x, t) :: e) ((x, t) :: e').$$

Lemma  $\text{has\_type\_weaken}$ :

$$\forall e \ a \ t, \text{has\_type } e \ a \ t \rightarrow \forall e', \text{wf\_env } e' \rightarrow \text{env\_weaken } e \ e' \rightarrow \text{has\_type } e' \ a \ t.$$

The  $\text{has\_type}$  predicate is equivariant, i.e. stable by swapping.

Lemma  $\text{has\_type\_swap}$ :

$$\begin{aligned}
& \forall u \ v, \text{kind } u = \text{kind } v \rightarrow \\
& \forall e \ a \ t, \text{has\_type } e \ a \ t \rightarrow \\
& \quad \text{has\_type } (\text{swap\_env } u \ v \ e) (\text{swap\_term } u \ v \ a) (\text{swap\_type } u \ v \ t).
\end{aligned}$$

As a consequence of equivariance, we obtain admissible typing rules for functions and type abstractions, similar to rules  $t\_abs$  and  $t\_tabs$  but where the variable name is quantified existentially rather than universally.

Lemma  $\text{kind\_fv\_type}$ :  $\forall e \ t, \text{wf\_type } e \ t \rightarrow \forall x, \text{In } x (\text{fv\_type } t) \rightarrow \text{kind } x = \text{TYPE}$ .

Lemma *fv\_wf\_type\_kind*:  $\forall x e t, wf\_type e t \rightarrow kind x = TERM \rightarrow \neg In x (fv\_type t)$ .

Lemma *fresh\_freshen\_term*:

$\forall x t1 e a y,$   
 $wf\_term ((x, t1) :: e) (freshen\_term a x) \rightarrow \neg In y (dom e) \rightarrow x \neq y \rightarrow$   
 $\neg In y (fv\_term a).$

Lemma *t\_abs'*:

$\forall e t1 a t2 x,$   
 $kind x = TERM \rightarrow \neg In x (dom e) \rightarrow \neg In x (fv\_term a) \rightarrow$   
 $has\_type ((x, t1) :: e) (freshen\_term a x) t2 \rightarrow$   
 $has\_type e (Fun t1 a) (Arrow t1 t2).$

Lemma *fresh\_freshen\_tety*:

$\forall x t1 e a y,$   
 $wf\_term ((x, t1) :: e) (freshen\_tety a x) \rightarrow \neg In y (dom e) \rightarrow x \neq y \rightarrow$   
 $\neg In y (fv\_term a).$

Lemma *t\_tabs'*:

$\forall e t1 a t2 x,$   
 $kind x = TYPE \rightarrow \neg In x (dom e) \rightarrow \neg In x (fv\_term a) \rightarrow \neg In x (fv\_type t2) \rightarrow$   
 $has\_type ((x, t1) :: e) (freshen\_tety a x) (freshen\_type t2 x) \rightarrow$   
 $has\_type e (TFun t1 a) (Forall t1 t2).$

### 3.3 Stability of the typing judgement under substitutions

We now show that the typing judgement is stable under substitutions. There are two substitutions to consider: of a type for a type variable, and of a term for a term variable.

Lemma *has\_type\_stable\_type\_subst*:

$\forall e1 x p q e2 a t,$   
 $kind x = TYPE \rightarrow$   
 $is\_subtype e1 p q \rightarrow$   
 $has\_type (e2 ++ (x, q) :: e1) a t \rightarrow$   
 $has\_type (psubst\_env e2 x p ++ e1) (psubst\_tety a x p) (psubst\_type t x p).$

Lemma *lookup\_env\_append*:

$\forall e2 x p y e1,$   
 $wf\_env (e1 ++ (x, p) :: e2) \rightarrow$   
 $lookup y (e1 ++ (x, p) :: e2) = if eq\_name y x then Some p else lookup y (e1 ++ e2).$

Lemma *wf\_env\_append*:

$\forall e2 x p e1, wf\_env (e1 ++ (x, p) :: e2) \rightarrow kind x = TERM \rightarrow wf\_env (e1 ++ e2).$

Lemma *is\_subtype\_strengthen*:

$$\begin{aligned} & \forall e s t, \text{is\_subtype } e s t \rightarrow \\ & \forall e', \text{wf\_env } e' \rightarrow (\forall x : \text{name}, \text{kind } x = \text{TYPE} \rightarrow \text{lookup } x e' = \text{lookup } x e) \rightarrow \\ & \text{is\_subtype } e' s t. \end{aligned}$$

Lemma *has\_type\_stable\_term\_subst*:

$$\begin{aligned} & \forall e1 x b s e2 a t, \\ & \text{kind } x = \text{TERM} \rightarrow \text{has\_type } e1 b s \rightarrow \text{has\_type } (e2 ++ (x, s) :: e1) a t \rightarrow \\ & \text{has\_type } (e2 ++ e1) (\text{psubst\_term } a x b) t. \end{aligned}$$

### 3.4 Dynamic semantics

The dynamic semantics of  $F_{<}$  is specified by a one-step reduction relation, in small-step operational style. We first define values (final results of reduction sequences) as a subset of terms.

Inductive *isvalue*: *term*  $\rightarrow$  *Prop* :=

$$\begin{aligned} & | \text{isvalue\_fun}: \forall t a, \\ & \quad \text{isvalue } (\text{Fun } t a) \\ & | \text{isvalue\_tfun}: \forall t a, \\ & \quad \text{isvalue } (\text{TFun } t a). \end{aligned}$$

We first give a Plotkin-style specification of the reduction relation: it uses inductive rules *red\_appfun*, *red\_apparg*, *red\_tapp* instead of contexts to describe reductions inside applications. The two rules *red\_appabs* and *red\_taptabs* are the familiar beta-reduction rules for term and type applications, respectively.

Inductive *red*: *term*  $\rightarrow$  *term*  $\rightarrow$  *Prop* :=

$$\begin{aligned} & | \text{red\_appabs}: \forall t a v, \\ & \quad \text{isvalue } v \rightarrow \\ & \quad \text{red } (\text{App } (\text{Fun } t a) v) (\text{vsubst\_term } a 0 v) \\ & | \text{red\_taptabs}: \forall t a t', \\ & \quad \text{red } (\text{TApp } (\text{TFun } t a) t') (\text{vsubst\_tety } a 0 t') \\ & | \text{red\_appfun}: \forall a a' b, \\ & \quad \text{red } a a' \rightarrow \text{red } (\text{App } a b) (\text{App } a' b) \\ & | \text{red\_apparg}: \forall v b b', \\ & \quad \text{isvalue } v \rightarrow \text{red } b b' \rightarrow \text{red } (\text{App } v b) (\text{App } v b') \\ & | \text{red\_tapp}: \forall a a' t, \\ & \quad \text{red } a a' \rightarrow \text{red } (\text{TApp } a t) (\text{TApp } a' t). \end{aligned}$$

We now give an alternate specification of the reduction relation in the style of Wright and Felleisen. The *red\_top* relation captures beta-reductions at the top of a term. Reductions within terms are expressed using reduction contexts (see the *red\_context* relation). Contexts are represented as functions from terms to terms whose shape is constrained by the *is\_context* predicate.



Inductive *red\_top*: *term*  $\rightarrow$  *term*  $\rightarrow$  *Prop* :=  
 | *red\_top\_appabs*:  $\forall t a v,$   
   *isvalue* *v*  $\rightarrow$   
   *red\_top* (*App* (*Fun* *t a*) *v*) (*vsubst\_term* *a* 0 *v*)  
 | *red\_top\_tapptabs*:  $\forall t a t',$   
   *red\_top* (*TApp* (*TFun* *t a*) *t'*) (*vsubst\_tety* *a* 0 *t'*).

Inductive *is\_context*: (*term*  $\rightarrow$  *term*)  $\rightarrow$  *Prop* :=  
 | *is\_context\_hole*:  
   *is\_context* (*fun* *a*  $\Rightarrow$  *a*)  
 | *is\_context\_app\_left*:  $\forall c b,$   
   *is\_context* *c*  $\rightarrow$  *is\_context* (*fun* *x*  $\Rightarrow$  *App* (*c x*) *b*)  
 | *is\_context\_app\_right*:  $\forall v c,$   
   *isvalue* *v*  $\rightarrow$  *is\_context* *c*  $\rightarrow$  *is\_context* (*fun* *x*  $\Rightarrow$  *App* *v* (*c x*))  
 | *is\_context\_tapp*:  $\forall c t,$   
   *is\_context* *c*  $\rightarrow$  *is\_context* (*fun* *x*  $\Rightarrow$  *TApp* (*c x*) *t*).

Inductive *red\_context*: *term*  $\rightarrow$  *term*  $\rightarrow$  *Prop* :=  
 | *red\_context\_intro*:  $\forall a a' c,$   
   *red\_top* *a a'*  $\rightarrow$  *is\_context* *c*  $\rightarrow$  *red\_context* (*c a*) (*c a'*).

The Plotkin-style relation is more convenient for doing formal proofs. Since the challenge is given in terms of contexts, we feel obliged to prove the equivalence between the two formulations of reduction. The proofs are routine inductions over the derivations of *red* and *is\_context*, respectively.

Lemma *red\_red\_context*:  $\forall a a',$  *red* *a a'*  $\rightarrow$  *red\_context* *a a'*.

Lemma *red\_context\_red*:  $\forall a a',$  *red\_context* *a a'*  $\rightarrow$  *red* *a a'*.

## 3.5 Type soundness proof

Type soundness for  $F_{<}$  is established by proving the standard properties of type preservation (also called subject reduction) and progress.

### 3.5.1 Preservation

Technical inversion lemmas on typing derivations. These lemmas are similar (but not fully identical) to lemma A.13 in the on-paper proof.

Lemma *has\_type\_fun\_inv*:  
 $\forall e a t,$  *has\_type* *e a t*  $\rightarrow$   
 $\forall b s1 u1 u2,$  *a = Fun* *s1 b*  $\rightarrow$  *is\_subtype* *e t* (*Arrow* *u1 u2*)  $\rightarrow$



```

rewrite (vsubst_psubst_freshen_type x); eauto.
apply t_sub with (psubst_type (freshen_type s2 x) x t0).
change e with (psubst_env nil x t0 ++ e).
apply has_type_stable_type_subst with t; eauto.
apply sub_trans with t1; auto.
simpl; auto.
change e with (psubst_env nil x t0 ++ e).
apply sub_stable_subst with t1; eauto. simpl; auto.
Case context left app
  apply t_app with t1; eauto.
Case context right app
  apply t_app with t1; eauto.
Case context left tapp
  apply t_tapp with t1; eauto.
Final conclusion
  eauto.
Qed.

```

### 3.5.2 Progress

The following lemma, which corresponds to lemma A.14 in the challenge statement, determines the shape of a value from its type. Namely, closed values of function types are function abstractions, and closed values of polymorphic types are type abstractions.

Lemma *canonical\_forms*:

```

∀ e a t, has_type e a t → e = nil → isvalue a →
match t with
| Arrow t1 t2 ⇒ ∃ s, ∃ b, a = Fun s b
| Forall t1 t2 ⇒ ∃ s, ∃ b, a = TFun s b
| Top ⇒ True
| _ ⇒ False
end.

```

The progress theorem shows that a term well-typed in the empty environment is never “stuck”: either it is a value, or it can reduce. The theorem is proved by a simple induction on the typing derivation for the term and a case analysis on whether the subterms of the term are values or can reduce further.

Theorem *progress*:  $\forall a t, \text{has\_type nil } a t \rightarrow \text{isvalue } a \vee \exists a', \text{red } a a'$ .

Proof.

```

assert (∀ e a t, has_type e a t → e = nil → isvalue a ∨ exists a', red a a').
induction 1; intros; subst e.

```

Free variable: impossible in the empty typing environment.

*simpl in H1. discriminate.*

Function: already a value.

*left; constructor.*

Application *App a b*.

*right.*

*destruct (IHhas\_type1 (refl\_equal \_)) as [Va | [a' Ra]].*

*destruct (IHhas\_type2 (refl\_equal \_)) as [Vb | [b' Rb]].*

*a* and *b* are values. *a* must be a *Fun*. Beta-reduction is possible.

*generalize (canonical\_forms - - - H (refl\_equal \_) Va).*

*intros [s [c EQ]]. subst a.*

*exists (vsubst\_term c 0 b). constructor. auto.*

*a* is a value, but *b* reduces. *App a b* therefore reduces.

*exists (App a b'). constructor; auto.*

*a* reduces. *App a b* reduces as well.

*exists (App a' b). constructor; auto.*

Type abstraction: already a value.

*left; constructor.*

Type application *TApp a t*.

*right. destruct (IHhas\_type (refl\_equal \_)) as [Va | [a' Ra]].*

*a* is a value. *a* must be a *TFun*. Beta-reduction is possible.

*generalize (canonical\_forms - - - H (refl\_equal \_) Va).*

*intros [s [b EQ]]. subst a.*

*exists (vsubst\_tety b 0 t). constructor.*

*a* reduces, and so does *TApp a t*.

*exists (TApp a' t). constructor; auto.*

Subtyping step.

*auto.*

Final conclusion.

*eauto.*

Qed.



## Chapter 4

# Execution of the dynamic semantics

In this chapter, we consider the problem of executing  $F_{<}$  terms as prescribed by the reduction semantics for this language. Such executions are useful for testing that the semantics has the intended behavior. This goal is listed as part 3 in the POPLmark challenge. As we will see, our development will go one step further and result in the production of an efficient and provably correct interpreter for  $F_{<}$ .

There are two approaches to executing dynamic semantics within Coq. The first operates directly on a relational specification of the semantics, either big-step or small-step like our *red* predicate from chapter 3. The *eauto* Coq tactic, which build proofs by Prolog-style resolution over a set of predeclared inference rules and lemmas, can be abused to search and build derivation trees for a goal of the form  $\exists b, \text{red } a \ b$ , therefore executing one reduction step from  $a$ . An example of this approach can be found in our work with A. Appel on the list-machine benchmark [AL06]. However, this approach is tricky to set up and very inefficient.

The other approach, which we follow in this chapter, is to specify the operational semantics as functions rather than predicates. While Coq has no efficient built-in execution mechanism for logic programs (composed of inductively-defined predicates), it can natively evaluate functional programs (composed of functions defined by recursion and pattern-matching). Such functional reductions are actually part of the logic of Coq, via the notion of conversion.

We therefore proceed in two steps. We will first define functions that compute the one-step or  $N$ -step reduct of a  $F_{<}$  term, and prove that they are correct and complete with respect to the relational semantics. We will then use these functions to evaluate terms within Coq and to extract efficient Caml code for an interpreter.

```
Require Import Arith.  
Require Import ZArith.
```

```

Require Import List.
Require Import extralibrary.
Require Import part1a.
Require Import part2a.

```

## 4.1 Execution of one-step reductions

We first show that the *isvalue* predicate is decidable. The lemma below will actually provides us with a decision procedure that takes any term *a* and returns whether it is a value or not. We can then use this decision procedure within function definitions.

Lemma *isvalue\_dec*:

$$\forall a, \{isvalue\ a\} + \{\sim isvalue\ a\}.$$

The *reduce* function maps a term *a* to either *Some b* if *a* reduces in one step to *b*, or to *None* if *a* does not reduce. It is defined by structural recursion over *a* and case analysis on whether subterms of *a* are values, or reduce, or are stuck.

Fixpoint *reduce* (*a*: term) : option term :=

```

  match a with
  | App b c =>
    if isvalue_dec b then
      if isvalue_dec c then
        match b with Fun t d => Some (vsubst_term d 0 c) | _ => None end
      else
        match reduce c with Some c' => Some(App b c') | None => None end
    else
      match reduce b with Some b' => Some(App b' c) | None => None end
  | TApp b t =>
    if isvalue_dec b then
      match b with TFun t' c => Some (vsubst_tety c 0 t) | _ => None end
    else
      match reduce b with Some b' => Some(TApp b' t) | None => None end
  | _ => None
end.

```

We then show that this function is correct and complete with respect to the reduction rules: *reduce a = Some b* if and only if *red a b* holds. The proofs are routine inductions on the structure of *a* for the “only if” part and on the derivation of *red a b* for the “if” part.

Lemma *reduce\_is\_correct*:

$$\forall a\ a',\ reduce\ a = Some\ a' \rightarrow red\ a\ a'.$$

Lemma *isvalue\_dec\_true*:

$$\forall a (T: \text{Set}) (b c: T), \text{isvalue } a \rightarrow (\text{if isvalue\_dec } a \text{ then } b \text{ else } c) = b.$$

Lemma *isvalue\_dec\_false*:

$$\forall a a' (T: \text{Set}) (b c: T), \text{red } a a' \rightarrow (\text{if isvalue\_dec } a \text{ then } b \text{ else } c) = c.$$

Lemma *reduce\_is\_complete*:

$$\forall a a', \text{red } a a' \rightarrow \text{reduce } a = \text{Some } a'.$$

## 4.2 Execution of $N$ -step reductions

The following function iterates the one-step reduction function *compute* to obtain the normal form of a term. Since Coq functions must always terminate, we need to bound the number of iterations by the  $n$  parameter. If a normal form cannot be reached in  $n$  steps, *compute* returns *None*.

```
Fixpoint compute (n: nat) (a: term) {struct n}: option term :=
  match n with
  | O => None
  | S n' =>
    match reduce a with
    | Some a' => compute n' a'
    | None => Some a
    end
  end.
```

We now show that *compute*  $a$ , if it succeeds, returns a reduct of  $a$  that is in normal form (irreducible).

Definition *irreducible* ( $a: \text{term}$ ): *Prop* :=  $\forall b, \neg \text{red } a b$ .

Inductive *red\_sequence*:  $\text{term} \rightarrow \text{term} \rightarrow \text{Prop}$  :=

```
| red_sequence_0:
   $\forall a, \text{irreducible } a \rightarrow \text{red\_sequence } a a$ 
| red_sequence_1:  $\forall a b c,$ 
   $\text{red } a b \rightarrow \text{red\_sequence } b c \rightarrow \text{red\_sequence } a c.$ 
```

Lemma *compute\_correct*:

$$\forall n a a', \text{compute } n a = \text{Some } a' \rightarrow \text{red\_sequence } a a'.$$

Conversely, if a term  $a$  has a normal form  $a'$ , there exists a number of iterations  $n$  such that *compute* returns *Some*  $a'$ .

Lemma *compute\_complete*:

$$\forall a a', \text{red\_sequence } a a' \rightarrow \exists n, \text{compute } n a = \text{Some } a'.$$



### 4.3 Experiments

We can now use the Coq directives *Eval compute in (reduce a)* and *Eval compute in (compute N a)* to display the results of performing one or  $N$  reduction steps in  $a$ .

Definition  $F\_poly\_identity := TFun Top (Fun (Tvar 0) (Var 0))$ .

Definition  $F\_top\_identity := TApp F\_poly\_identity Top$ .

Definition  $F\_delta := Fun (Arrow Top Top) (App (Var 0) (Var 0))$ .

Definition  $F\_testprog := App F\_delta F\_top\_identity$ .

*Eval compute in (reduce F\_testprog)*.

*Eval compute in (compute 100 F\_testprog)*.

The latter returns *Some (Fun Top (Var 0))*, which is indeed the value of the term  $F\_testprog$ . For a larger example, here is some arithmetic on Church integers.

Definition  $F\_one : term :=$

$$(TFun Top (TFun (Tvar 0) (TFun (Tvar 1) (Fun (Arrow (Tvar 2) (Tvar 1)) (Fun (Tvar 0) (App (Var 1) (Var 0))))))))).$$

Definition  $F\_nat : type :=$

$$(Forall Top (Forall (Tvar 0) (Forall (Tvar 1) (Arrow (Arrow (Tvar 2) (Tvar 1)) (Arrow (Tvar 0) (Tvar 1)))))).$$

Definition  $F\_add : term :=$

$$(Fun F\_nat (Fun F\_nat (TFun Top (TFun (Tvar 0) (TFun (Tvar 1) (Fun (Arrow (Tvar 2) (Tvar 1)) (Fun (Tvar 0) (App (TApp (TApp (TApp (Var 3) (Tvar 2)) (Tvar 1)) (Tvar 1)) (App (Var 1) (App (TApp (TApp (TApp (Var 2) (Tvar 2)) (Tvar 1)) (Tvar 0)) (App (Var 1) (Var 0)))))))))))))).$$

*Eval compute in (compute 100 (App (App F\_add F\_one) F\_one))*.

Execution is nearly instantaneous. In Coq 8.1, we can also use *Eval vm\_compute* to request evaluation via compilation to virtual machine code. This results in execution speed comparable to that of bytecoded OCaml.

An alternate execution path is to generate (or “extract” in Coq’s terminology) Caml code from the Coq definition of function *compute*. This is achieved by the following command:

```
Extraction "/tmp/fsub_eval.ml" compute.
```

The generated Caml code can be compiled with the OCaml native-code compiler for even higher execution speed. More importantly, it can be linked with a lexer, parser and printer hand-written in OCaml, obtaining a stand-alone reference interpreter for  $F_{<}$ , that can execute non-trivial programs.



## Chapter 5

# Assessment

The POPLmark challenge is still ongoing, and the many solutions submitted differ widely — in the encodings of binders used, in the proof assistants used, and even in the proof styles of each author. It is therefore too early to draw conclusions. In this final chapter, we try to assess the quality of our solution with respect to various criteria.

**Legibility of definitions and theorems** Mechanized proofs often define notions and state theorems in ways that are somewhat different from what a mathematician would do in an on-paper proof. Often, the mechanized definitions and statements are more precise, but also harder to read and to relate with one’s intuitions. The locally nameless approach followed in this report remains quite close to the on-paper definitions and theorems given in the statement of the POPLmark challenge. Definitions and statements are mostly unsurprising, except perhaps for the cases of definitions that involve crossing a binder. For instance, definition of subtyping between  $\forall$  types is written on paper as

$$\frac{\Gamma \vdash \tau_1 <: \sigma_1 \quad \Gamma, X <: \tau_1 \vdash \sigma_2 <: \tau_2}{\Gamma \vdash (\forall X <: \sigma_1. \sigma_2) <: (\forall X <: \tau_1. \tau_2)}$$

while in our approach it is written as

$$\frac{\Gamma \vdash \tau_1 <: \sigma_1 \quad \forall X, X \notin \text{Dom}(\Gamma) \implies \Gamma, X <: \tau_1 \vdash \sigma_2[0 \leftarrow X] <: \tau_2[0 \leftarrow X]}{\Gamma \vdash (\forall <: \sigma_1. \sigma_2) <: (\forall <: \tau_1. \tau_2)}$$

There are several “tricks” here that surprise the reader: the de Bruijn notation for the bound variable, the substitutions  $[0 \leftarrow X]$  in the second premise, and the placement of the quantification over  $X$ . In contrast, a purely nominal approach such as the solution by Urban *et al.* leads to a less mysterious definition of the form

$$\frac{\Gamma \vdash \tau_1 <: \sigma_1 \quad X \text{ fresh for } \Gamma \quad \Gamma, X <: \tau_1 \vdash \sigma_2 <: \tau_2}{\Gamma \vdash (\forall X <: \sigma_1. \sigma_2) <: (\forall X <: \tau_1. \tau_2)}$$

Nonetheless, we believe that definitions and theorems written in the locally nameless style are globally more readable and intuitively understandable than their equivalents using either de Bruijn indices or higher-order abstract syntax.

**Overheads** Mechanized proofs must make explicit a great number of small details and obvious properties that are omitted in research papers. In the case of type systems and operational semantics, this includes basic properties of terms, binders ( $\alpha$ -conversion), substitutions, typing environments, and reduction sequences. All these properties had to be painfully made explicit in the present development. We estimate that such scaffolding work represent more than half of our development. More precisely, we consider that the following parts are pure overhead, compared with a paper proof:

- Definition and properties of free variables.
- Definition and properties of substitutions. There are two substitutions per data type of interest (one for free names and the other for bound de Bruijn variables), with the associated commutation properties.
- Definition and properties of swaps.
- Proving equivariance for many definitions.
- Proving admissibility of the rules with  $\exists X \dots$  in the premises, instead of  $\forall X \dots$ .
- Manipulations of environments, e.g. properties of environments of the form  $\Gamma_1, X <: \tau, \Gamma_2$ .

A legitimate question to ask is: how much of this overhead could be avoided, either by factoring the definitions and properties in reusable libraries, or by generating them automatically from high-level descriptions of the syntax of terms? Examples of reusable libraries include Urban and Tasson’s nominal package, based on Isabelle/HOL type classes [UT05], and Chlipala’s library for typing environments. However, such forms of reuse are limited in Coq by lack of type classes and any other form of “polytypic” definitions. Automatic generation in the style of Pottier’s Caml [Pot06] is another direction that remains to be investigated.

**Size of the development** Our solution is neither particularly compact nor excessively verbose compared with other solutions. Arthur Charguéraud compared the sizes of several Coq solutions to part 1A of the challenge, counting the number of non-trivial tactics invoked. The results are as follows:

Authors	Tactics	Representation used
J�erome Vouillon	431	de Bruijn indices
Aaron Stump	1147	names and levels
Xavier Leroy	630	locally nameless
Hirschowitz & Maggesi	1615	de Bruijn indices
Adam Chlipala	342	locally nameless
Arthur Chargu�eraud	233	locally nameless

Several factors contribute to this relative verbosity of our solution. First, as we discussed above, the locally nameless approach comes with inherent overheads. Second, as pointed out by Chargu eraud, there are small variations of the locally nameless approach that can reduce the proof effort significantly (see below). Finally, unlike Vouillon, Chlipala and Chargu eraud, we did not really take advantage of Coq’s facilities for defining domain-specific tactics.

**Possible improvements to the locally nameless approach** Chargu eraud [Cha06] and Chargu eraud, Pierce and Weirich [CPW06] present several variations on the locally nameless approach that could significantly simplify our development. The most important one is to define predicates in the binder-crossing by quantification over all names not in a given finite set  $L$  of names, rather than all fresh names:

$$\forall L, \frac{\Gamma \vdash \tau_1 <: \sigma_1 \quad \forall X, X \notin L \implies \Gamma, X <: \tau_1 \vdash \sigma_2[0 \leftarrow X] <: \tau_2[0 \leftarrow X]}{\Gamma \vdash (\forall <: \sigma_1. \sigma_2) <: (\forall <: \tau_1. \tau_2)}$$

With this simple device, it is no longer necessary to prove equivariance of the well-formedness and subtyping relations. In turn, this removes the need to define and reason about swaps of names. The downside of this approach is reduced legibility: it becomes even less obvious why this rule captures the correct notion of subtyping between quantified types.

The second simplification is to treat well-formed typing environments as unordered sets of bindings, rather than ordered lists. For instance, instead of reasoning over environments of the form  $\Gamma_1, \Gamma_2$ , it is easier and often sufficient to reason over environments  $\Gamma$  that contain all bindings from  $\Gamma_1$  and all bindings from  $\Gamma_2$ . Similarly, well-formedness of terms and types can profitably be defined with respect to a set of names bound in an environment rather than with respect to an environment.

Finally, the definition of the `vsubst` substitutions (replacement of a bound variable by a term) can also be simplified: instead of taking

$$\text{vsubst\_type } (\text{Tvar } n) \ x \ b = \begin{cases} \text{Tvar } n, & \text{if } n < x \\ b, & \text{if } n = x \\ \text{Tvar } (n - 1), & \text{if } n > x \end{cases}$$

it suffices to take

$$\text{vsubst\_type } (\text{Tvar } n) \ x \ b = \begin{cases} \text{Tvar } n, & \text{if } n \neq x \\ b, & \text{if } n = x \end{cases}$$

since the case  $n > x$  is never exercised in the rest of the development. This simple change significantly simplifies the proofs of the commutation lemmas between `psubst` and `vsubst` substitutions.

**Executability of the semantics** Part 3 of the POPLmark challenge (testing and animating the dynamic semantics) did not receive much attention from the participants. However, we believe that it is generally important to generate correct-by-construction implementations of programming tools from high-level specifications. The Coq proof assistant takes an almost schizophrenic stance on this issue. On the one hand, Coq provides excellent execution facilities for definitions written in functional style, both via its built-in evaluator (using an efficient virtual machine since version 8.1) and via its code extraction facility. In other work [Ler06, BDL06], we developed a whole optimizing C compiler by extraction from Coq functional specifications. On the other hand, Coq supports poorly the execution of specifications written in relational style (inductive predicates), forcing users to manually write functional variants of their relational specifications and to prove the equivalence of the functional and relational specifications, as we did in chapter 4. This is an annoyance in general, but also an interesting feature in some cases. In particular, it encourages a style where relational specifications are not polluted by executability constraints, and can e.g. be non-deterministic or even undecidable; executability constraints can then be taken into account later, in typical refinement style, when developing the functional specification.

**Extension to records and record types** The POPLmark challenge suggests to extend system  $F_{<}$  with records and record types with depth and width subtyping (parts 1B and 2B of the challenge). We considered briefly this extension but did not pursue it. This extension is conceptually rather easy but technically difficult: as in the other solutions that addressed parts 1B and 2B of the challenge, it appears necessary to define types and record types, as well as terms and record terms, in a mutually recursive manner. Such mutually recursive definitions are quite painful to handle in Coq, and require extensive changes to the proof scripts. Again, this is more a limitation of Coq than a fundamental difficulty.

# Bibliography

- [ABF<sup>+</sup>05] Brian E. Aydemir, Aaron Bohannon, Matthew Fairbairn, J. Nathan Foster, Benjamin C. Pierce, Peter Sewell, Dimitrios Vytiniotis, Geoffrey Washburn, Stephanie Weirich, and Steve Zdancewic. Mechanized metatheory for the masses: The POPLmark challenge. In *International Conference on Theorem Proving in Higher Order Logics (TPHOLs)*, volume 3603 of *Lecture Notes in Computer Science*, pages 50–65. Springer-Verlag, 2005.
- [AL06] Andrew W. Appel and Xavier Leroy. A list-machine benchmark for mechanized metatheory. Research report 5914, INRIA, 2006.
- [BC04] Yves Bertot and Pierre Castéran. *Interactive Theorem Proving and Program Development – Coq’Art: The Calculus of Inductive Constructions*. EATCS Texts in Theoretical Computer Science. Springer-Verlag, 2004.
- [BDL06] Sandrine Blazy, Zaynah Dargaye, and Xavier Leroy. Formal verification of a C compiler front-end. In *FM 2006: Int. Symp. on Formal Methods*, volume 4085 of *Lecture Notes in Computer Science*, pages 460–475. Springer-Verlag, 2006.
- [Cha06] Arthur Charguéraud. A comparison between concrete representations for bindings. In *Workshop on Mechanizing Metatheory*, 2006.
- [Coq07] The Coq proof assistant. Software and documentation available on the Web, <http://coq.inria.fr/>, 1995–2007.
- [CPW06] Arthur Charguéraud, Benjamin C. Pierce, and Stephanie Weirich. Proof engineering: Practical techniques for mechanized metatheory. Manuscript, available from <http://www.cis.upenn.edu/~bcpierce/papers/binders.pdf>, September 2006.
- [Gor94] Andrew D. Gordon. A mechanisation of name-carrying syntax up to alpha-conversion. In *Higher-order logic theorem proving and its applications 1993*, volume 780 of *Lecture Notes in Computer Science*, pages 414–426. Springer-Verlag, 1994.



- 
- [Hue94] Gérard Huet. Residual theory in lambda-calculus: A formal development. *Journal of Functional Programming*, 4(3):371–394, 1994.
- [Ler06] Xavier Leroy. Formal certification of a compiler back-end, or: programming a compiler with a proof assistant. In *33rd symposium Principles of Programming Languages*, pages 42–54. ACM Press, 2006.
- [Ler07] Xavier Leroy. A locally nameless solution to the POPLmark challenge – the Coq development. <http://gallium.inria.fr/~xleroy/POPLmark/locally-nameless>, 2007.
- [MM04] Conor McBride and James McKinna. I am not a number; I am a free variable. In *Proc. 2004 Haskell Workshop*. ACM Press, 2004.
- [MP99] James McKinna and Randy Pollack. Some lambda calculus and type theory formalized. *J. Automated Reasoning*, 23:373–409, 1999.
- [Pit03] A. M. Pitts. Nominal logic, a first order theory of names and binding. *Information and Computation*, 186:165–193, 2003.
- [Pot06] François Pottier. An overview of Caml. In *ACM Workshop on ML*, volume 148 of *Electronic Notes in Theoretical Computer Science*, pages 27–52, March 2006.
- [UT05] Christian Urban and Christine Tasson. Nominal reasoning techniques in Isabelle/HOL. In *Proc. Int. Conf. on Automated Deduction (CADE)*, volume 3632 of *Lecture Notes in Computer Science*, pages 38–53. Springer-Verlag, 2005.



---

Unité de recherche INRIA Rocquencourt  
Domaine de Voluceau - Rocquencourt - BP 105 - 78153 Le Chesnay Cedex (France)

Unité de recherche INRIA Futurs : Parc Club Orsay Université - ZAC des Vignes  
4, rue Jacques Monod - 91893 ORSAY Cedex (France)

Unité de recherche INRIA Lorraine : LORIA, Technopôle de Nancy-Brabois - Campus scientifique  
615, rue du Jardin Botanique - BP 101 - 54602 Villers-lès-Nancy Cedex (France)

Unité de recherche INRIA Rennes : IRISA, Campus universitaire de Beaulieu - 35042 Rennes Cedex (France)

Unité de recherche INRIA Rhône-Alpes : 655, avenue de l'Europe - 38334 Montbonnot Saint-Ismier (France)

Unité de recherche INRIA Sophia Antipolis : 2004, route des Lucioles - BP 93 - 06902 Sophia Antipolis Cedex (France)

---

Éditeur  
INRIA - Domaine de Voluceau - Rocquencourt, BP 105 - 78153 Le Chesnay Cedex (France)  
<http://www.inria.fr>  
ISSN 0249-6399